

# TERRAN BASIC REFERENCE MANUAL

For Language Version 1.2 | Third Edition

© 2020– Minjae Song (“CuriousTorvald”)

Copyrighted under the terms of MIT License

**O'REALLY<sup>2</sup>** Press, Cyberworld

First Edition (for version 1.0):	2020-12-28
Second Edition (for version 1.1):	2021-01-28
Second Impression, with minor corrections:	2021-02-01
Third Edition (for version 1.2):	2021-05-05

# Contents

<b>1</b>	<b>Introduction</b>	<b>7</b>
<b>2</b>	<b>Version Changes</b>	<b>8</b>
2.1	Changes from 1.1 . . . . .	8
2.2	Changes from 1.0 . . . . .	8
<b>I</b>	<b>Language</b>	<b>9</b>
<b>3</b>	<b>Basic Concepts</b>	<b>10</b>
3.1	Values and Types . . . . .	10
3.2	Control Flow . . . . .	11
<b>4</b>	<b>Language Guide</b>	<b>12</b>
4.1	GOTO . . . . .	12
4.2	Subroutine with GOSUB . . . . .	13
4.3	FOR–NEXT Loop . . . . .	13
4.4	Get User INPUT . . . . .	14
4.5	Array . . . . .	14
4.6	Function . . . . .	15
4.7	Recursion . . . . .	15
4.8	When Recursion Goes Wild . . . . .	16
4.9	Higher-order Function . . . . .	17
4.10	MAPping . . . . .	17
4.11	Closure . . . . .	18
4.12	Currying . . . . .	18
4.13	Example: Quicksort . . . . .	19
4.14	Monad . . . . .	20
<b>5</b>	<b>Language Reference</b>	<b>21</b>
5.1	Metasyntax . . . . .	21
5.2	Definitions . . . . .	21
5.3	Literals . . . . .	22
5.4	Operators . . . . .	23
5.5	Constants . . . . .	28
5.6	Syntax In EBNF . . . . .	28
<b>6</b>	<b>Commands</b>	<b>31</b>
6.1	The Editor . . . . .	31

<b>7</b>	<b>Statements</b>	<b>34</b>
7.1	IF . . . . .	34
7.2	ON . . . . .	34
7.3	DEFUN . . . . .	35
<b>8</b>	<b>Functions</b>	<b>36</b>
8.1	Mathematical . . . . .	36
8.2	Input . . . . .	39
8.3	Output . . . . .	40
8.4	Program Manipulation . . . . .	41
8.5	String Manipulation . . . . .	43
8.6	Array Manipulation . . . . .	43
8.7	Monad Manipulation . . . . .	44
8.8	Graphics . . . . .	44
8.9	Meta . . . . .	45
8.10	System . . . . .	46
8.11	Higher-order Function . . . . .	46
<b>II</b>	<b>Implementation</b>	<b>48</b>
<b>9</b>	<b>Interpreter</b>	<b>49</b>
9.1	Resolving Variables . . . . .	49
9.2	Unresolved Values . . . . .	49
9.3	Lambda Variables . . . . .	50
<b>10</b>	<b>Virtual Machine</b>	<b>53</b>
10.1	Keycodes . . . . .	53
10.2	Code Page . . . . .	54
10.3	Colour Palette . . . . .	55
10.4	MMIO . . . . .	57
<b>III</b>	<b>More Goodies</b>	<b>58</b>
<b>11</b>	<b>99 Bottles of Beer</b>	<b>59</b>
<b>12</b>	<b>Amazing</b>	<b>60</b>
<b>13</b>	<b>Hamurabi</b>	<b>64</b>
<b>14</b>	<b>Hangman</b>	<b>68</b>

<i>CONTENTS</i>	5
<b>15 Plotter</b>	<b>71</b>
<b>16 Proof That Monad Laws Are Obeyed</b>	<b>72</b>
<b>17 Bibliography</b>	<b>74</b>
<b>18 Copyright</b>	<b>75</b>



# Chapter 1

## Introduction

Terran BASIC is a BASIC dialect and its interpreter. Terran BASIC emulates most of the common BASIC syntax while also adding more advanced and up-to-date concepts gracefully, such as a user-defined function that can *actually* recurse, arbitrary list construction using CONS-operator and some of the features in the realm of functional programming from Map and Fold to Closure and Currying.

This is the documentation for Terran BASIC 1.2.

*Henceforward this documentation will use more friendly and sometimes vulgar language because that's more fun to write!*

# Chapter 2

## Version Changes

### 2.1 Changes from 1.1

- Construct an arbitrary array using { and }
- Three new functor operators: <\$> , <~> and <\*>
- One new function operator: &
- One new graphics function: CLPX
- Trying to print an array will print them formatted with all the curly braces
- Various fixes to the TSVM, notably the graphics part

### 2.2 Changes from 1.0

- Adding support for anonymous function (closure)
- The editor can now delete program lines
- The editor will warn you of overwriting when you try to load a basic program
- Adding two new function operators: \$ and .
- Adding a monad to the type system and four monad operators: >>= >>~ \ @ , and two monad functions: MJOIN and MRET
- Adding an undefined to the type system
- Adding two new constants: ID and UNDEFINED
- Built-in functions can be used on built-in higher-order functions, namely MAP, FILTER and FOLD
- GOTOYX function to move the text cursor
- Implementation: definition of the Lambda Variable Index has changed, in which its ordIndex is no longer in reverse
- Doc: added documentation for TSVM code page, colour palette and MMIO
- Fix: FOR would not work with non-integer numbers
- Fix: PLOT would not work at all



## **Part I**

# **Language**

# Chapter 3

## Basic Concepts

*“Caution! Under no circumstances confuse the adjective basic with the noun BASIC, except under confusing circumstances!”*

— Terran BASIC Reference Manual, Third Edition\*

This chapter describes the basic concepts of the Terran BASIC language.

### 3.1 Values and Types

BASIC is a *Dynamically Typed Language*, which means variables do not know which group they should barge in; only values of the variable do. In fact, there is no type definition in the language: *we do want our variables to feel awkward*.

There are eight basic types: *number*, *boolean*, *string*, *array*, *generator*, *function*, *monad* and *undefined*.

*Number* represents real (double-precision floating-point or actually, *rational*) numbers. Operations on numbers follow the same rules of the underlying virtual machine<sup>†</sup>, and such machines must follow the IEEE 754 standard<sup>‡</sup>.

*Boolean* is type of the value that is either **TRUE** or **FALSE**. Number **0** and **FALSE** make the condition *false*. When used in numeric context, **FALSE** will be interpreted as 0 and **TRUE** as 1.

*String* represents immutable<sup>§</sup> sequences of bytes.

*Array* represents a collection of *things*<sup>¶</sup> in one or more dimensions.

*Generator* represents a value that automatically counts up/down whenever they have been called in FOR–NEXT loop.

*Function* is, well... a function, especially user-defined one. Functions are *type* because some built-in functions will actually take *functions* as arguments.

---

\*Original quotation: Donald R. Woods, James M. Lyon, *The INTERCAL Programming Language Reference Manual*

<sup>†</sup>If you are not a computer person, disregard.

<sup>‡</sup>ditto.

<sup>§</sup>Cannot be altered directly.

<sup>¶</sup>Arrays can have any numbers, booleans, strings, arrays (but not itself), generators, functions, monads and undefined.

*Monad* is a type that contains some value and follows monad laws. The term Monad refers to any object that satisfies these requirements, however, in Terran BASIC, only one monad type does that (and is useful to you): value-monad.

*Undefined* represents a value that holds nothing. A function's unspecified arguments, when examined, have this type and are equal to `UNDEFINED`. User-defined functions are free to have `UNDEFINED` as its arguments, but only the highly limited set of built-in functions will accept one.

### 3.1.1 Supertypes

Some types are so similar, it is helpful to classify them into one greater type.

*Functor* is a collection of values that can be `MAP` ped. Functors include Arrays and Generators.

## 3.2 Control Flow

A program is executed starting with its lowest line number. Statements on a line are executed from left to right. When all Statements are finished execution, the next lowest line number will be executed. Control flow functions can modify this normal flow.

You can dive into other lines in the middle of the program with `GOTO`. The program flow will continue normally at the new line *and it will never know ya just did that*.

If you want less insane jumping, `GOSUB` is used to jump to a subroutine. A subroutine is a little section of a code that serves as a tiny program inside of a program. `GOSUB` will remember from which statement in the line you have come from, and will return your program flow to that line when `RETURN` statement is encountered. (of course, if `RETURN` is used without `GOSUB`, the program will raise some error) Do note that while you can reserve some portion of a program line as a `subroutine`, Terran BASIC does not provide local variables and whatnot<sup>||</sup> as all variables in BASIC are global, and you can just `GOTO` out of a subroutine to anywhere you desire and wreak havoc *if you really wanted to*.

The `ON` statement provides an alternative branching construct. You can enter multiple line numbers and let your variable (or mathematical expression) to choose which index of line numbers to `GOTO` - or `GOSUB` into.

The `IF-THEN-ELSE` lets you to conditionally select which of the two branches to execute.

The `FOR-NEXT` lets you loop a portion of a program while automatically counting your chosen variable up or down.

---

<sup>||</sup> If you really need some local variables, use the black magic of Monad.

# Chapter 4

## Language Guide

*“Begin at the beginning”, the King said gravely, “and go on till you come to the end: then stop.”*

— Lewis Carroll, *Alice in Wonderland*

We'll begin at the beginning; how beginning? This:

---

```
10 PRINT 2 + 2
run
4
Ok
```

---

Oh *boy* we just did a computation! It printed out `4` which is a correct answer for  $2 + 2$  and it didn't crash!

### 4.1 GOTO here and there

`GOTO` is used a lot in BASIC, and so does in Terran BASIC. `GOTO` is the simplest method of diverging a program flow: execute only the part of the program conditionally and perform a loop.

Following program attempts to calculate a square root of the input value, showing how `GOTO` can be used in such a manner.

---

```
10 X = 1337
20 Y = 0.5 * X
30 Z = Y
40 Y = Y - ((Y ^ 2) - X) / (2 * Y)
50 IF NOT(Z == Y) THEN GOTO 30 : REM 'NOT(Z==Y)' can be rewritten to
    'Z<>Y'
100 PRINT "Square root of ";X;" is approximately ";Y
```

---

Here, `GOTO` in line 50 is used to perform a loop, which keeps looping until `Z` and `Y` becomes equal. This is a Newtonian method of approximating a square root.

## 4.2 What If We Wanted to Go Back?

But `GOTO` only jumps, you can't jump *back*. Well, not with that attitude; you *can* go back with `GOSUB` and `RETURN` statement.

This program will draw a triangle, where the actual drawing part is on line 100–160, and only get jumped into it when needed.

---

```

10 GOTO 1000
100 REM subroutine to draw a segment. Size is stored to 'Q'
110 PRINT SPC(20 - Q);
120 Q1 = 1 : REM loop counter for this subroutine
130 PRINT "*";
140 Q1 = Q1 + 1
150 IF (Q1 <= Q * 2 - 1) THEN GOTO 130
160 PRINT : RETURN : REM this line will take us back from the jump
1000 Q = 1 : REM this is our loop counter
1010 GOSUB 100
1020 Q = Q + 1
1030 IF (Q <= 20) THEN GOTO 1010

```

---

## 4.3 FOR ever loop NEXT

As we've just seen, you can make loops using `GOTO`s here and there, but they *totally suck*, too much spaghetti crashes your cerebral cortex faster than *Crash Bandicoot 2*. Fortunately, there's a better way to go about that: the `FOR`–`NEXT` loop!

---

```

10 GOTO 1000
100 REM subroutine to draw a segment. Size is stored to 'Q'
110 PRINT SPC(20 - Q);
120 FOR Q1 = 1 TO Q * 2 - 1
130 PRINT "*";
140 NEXT : PRINT
150 RETURN
1000 FOR Q = 1 TO 20
1010 GOSUB 100
1020 NEXT

```

---

When executed, this program print out *exactly the same* triangle, but code is much more straightforward thanks to the `FOR` statement.

## 4.4 Isn't It Nice To Have a Computer That Will Question You?

What fun is the program if it won't talk with you? You can make that happen with `INPUT` statement.

---

```
10 PRINT "WHAT IS YOUR NAME";
20 INPUT NAME
30 PRINT "HELLO, ";NAME
```

---

This short program will ask your name, and then it will greet you by the name you told the computer.

## 4.5 Array

One variable is great, but having to use multiple variable to store multiple values related to each other? That totally sucks; fortunately there is a way to pack those values into one variable: the Array!

Consider the following example:

---

```
10 A = {2,3,5,7,11,13,17}
100 FOR K = 0 TO LEN(A) - 1
110 PRINT A(K)
120 NEXT
200 FOREACH AK IN A
210 PRINT AK
220 NEXT
```

---

An array is declared in the line 10, and line 100–120 and line 200–220 iterates the array to print it.

Line 200 uses `FOREACH` statement, which is same as `FOR` but made to work with arrays.

And can you put arrays inside of the array? Of course you can! And, *ahem*, let's consider the following example:

---

```
10 A = {{ "0", "X", "0" }, { "X", "0", "X" }, { "X", "X", "0" }}
20 FOR Y = 0 TO LEN(A) - 1
30   FOR X = 0 TO LEN(A(Y)) - 1
40     PRINT(A(Y, X); " ");
50   NEXT
```

---

```
60 PRINT
70 NEXT
```

---

The array `A` contains three arrays (making them multi-dimensional) and we're accessing them using this order: `A(outer index, inner index)`

## 4.6 Function

While you can put some pieces of codes some corner of the entire program and `GO-SUB` there, they're honestly bad — and you also have to keep track of which variables are used to hold temporary values, and there are more things that are just *bunch of pooppy nonsense*.

Consider the following code:

---

```
10 DEFUN POW2(N) = 2 ^ N
20 DEFUN DCOS(N) = COS(PI * N / 180)
30 FOR X = 0 TO 8
40 PRINT X,POW2(X)
50 NEXT
60 PRINT "-----"
70 FOREACH A IN {0,45,90,135,180}
80 PRINT A,DCOS(A)
90 NEXT
```

---

Here, we have defined two functions to use in the program: `POW2` and `DCOS`. Also observe that functions are defined using variable `N`s, but we use them with `X` in line 40 and with `A` in line 80: yes, functions can have their local name so you don't have to carefully choose which variable name to use in your subroutine.

Except a function can't have statements that span two- or more BASIC lines; but there are ways to get around that, including `DO` statement and *functional currying*.

## 4.7 Recursion

Functions can call themselves, and we call it ~~inception~~ recursion.

But why would we want to do that? Well, for one we can't use `FOR` statements within a function, but recursion sometimes makes certain problems easier.

Let's say we are counting a length of a repeating characters in a string, say "aaaabb-bccaxzyzy". This problem can be solved using a recursion like this:

---





---

Since `IF-THEN-ELSE` can be chained to make third or more conditions — `IF-THEN-ELSE IF-THEN` or something — we can write a recursive Fibonacci function:

---

```
10 DEFUN FIB(N) = IF (N == 0) THEN 0 ELSE IF (N == 1) THEN 1 ELSE
    FIB(N - 1) + FIB(N - 2)
20 FOR K = 1 TO 10
30 PRINT FIB(K); " ";
40 NEXT
```

---

## 4.9 The Functions of the High Order

Higher-order functions are functions that either takes another function as an argument, or returns a function. This sample program shows how higher-order functions can be constructed.

---

```
10 DEFUN APPLY(F,X) = F(X)
20 DEFUN FUN(X) = X ^ 2
30 K = APPLY(FUN, 42)
40 PRINT K
```

---

Here, `APPLY` takes a function `F` and value `X`, *applies* a function `F` onto the value `X` and returns the value. Since `APPLY` takes a function, it's higher-order function.

## 4.10 Map

`MAP` is a higher-order function that takes a function (called *transformation*) and an array to construct a new array that contains old array transformed with given *transformation*.

Or, think about the old `FAC` program before: it merely printed out the value of 1!, 2! ... 10!. What if we wanted to build an array that contains such values?

---

```
10 DEFUN FAC(N) = IF (N == 0) THEN 1 ELSE N * FAC(N - 1)
20 K = MAP(FAC, 1 TO 10)
30 PRINT K
```

---

Here, `K` holds the values of 1!, 2! ... 10!. Right now we're just printing out the array, but being an array, you can make actual use of it.

## 4.11 The Function with No Name

But `DEFUN F(X)` is only there for partial compatibility with traditional BASICs, of which their syntax is `DEF FNF(X)`. `DEFUN` cannot define nested functions, it's not a lambda-calculus system, yaddi yadda.

No, we want to be up-to-date; we don't always want every function to be global; we want to utter *give me a closure, bar-tender*.

Terran BASIC presents you: a closure. What does it do?

---

```
10 FAC = [N] ~> IF (N == 0) THEN 1 ELSE N * FAC(N - 1)
20 K = MAP(FAC, 1 TO 10)
30 PRINT K
```

---

Here, `[N] ~> ...` defines a *closure* (anonymous function) that has single parameter `N`, then assigns it to global variable `FAC`.

But *stop right there, criminal scum*: in what way is that an *anonymous function*?

Ah-ha, take a look at this:

---

```
10 F = [X] ~> MAP([X] ~> X <= 5, X)
20 PRINT F(1 TO 10)
```

---

Here, `MAP` inside of the global function `F` has an internal function: `[X] ~> X <= 5`

This function is anonymous: only the `MAP` can use it and is not accessible from the other scopes. Even if the `F` and the anonymous function use same parameter name of `X`, they don't matter because two functions are different.

## 4.12 Haskell Curry Wants to Know Your Location

Just three pages ago there was a mentioning about something called *functional currying*. So what the fsck is currying? Consider the following code:

---

```
10 DEFUN F(K,T) = ABS(T) == K
20 CF = F ~< 32
30 PRINT CF(24) : REM will print 'false'
40 PRINT CF(-32) : REM will print 'true'
```

---

Here, `CF` is a curried function of `F`; built-in operator `~<` applies `32` to the first pa-

parameter of the function `F`, which dynamically returns a *function* of `CF(T) = ABS(T) == 32`. The fact that Curry Operator returns a *function* opens many possibilities, for example, you can create loads of sibling functions without making loads of duplicate codes.

But, what if we pre-cook the curry before serve? The `~<` operator is there to curry an un-curried function; we wouldn't really need that if the function was curried in the begin with.

Here, *closure* do wonders as well; for a fun of it, let's re-write the `F(K, T)` to be pre-curried:

---

```
10 F = [K] ~> [T] ~> ABS(T) == K
20 CF = F(32)
30 PRINT CF(24) : REM will print 'false'
40 PRINT CF(-32) : REM will print 'true'
```

---

The function `F`, when called, returns its inner function `[T] ~> ABS(T) == K` with `K` being substituted with the argument that were applied to `F`, so the function `CF` here can be expressed as: `[T] ~> (T) == 32`

The subsequent calls for `CF` return appropriate values, in the same manner as the descriptions above.

## 4.13 Example: Quicksort

Using all the knowledge we have learned so far, it should be trivial\* to write a Quicksort function in Terran BASIC, like this:

---

```
10 QSORT = [XS] ~> IF (LEN(XS) < 1) THEN NIL ELSE
    QSORT(FILTER([X] ~> X < HEAD XS, TAIL XS)) # {HEAD XS} #
    QSORT(FILTER([X] ~> X >= HEAD XS, TAIL XS))
100 L={7,9,4,5,2,3,1,8,6}
110 PRINT L
120 PRINT QSORT(L)
```

---

Line 10 implements quicksort algorithm. `QSORT` selects a pivot by taking the head-element of the array `XS`. with `HEAD XS`, then utilises anonymous functions `[X] ~> X < HEAD XS` and `[X] ~> X >= HEAD XS` to move lesser-than-pivot values to the left and greater to the right (the head element itself does not get recursed, here `TAIL XS` is applied to make head-less copy of the array), and these two separated *chunks* are

---

\*/s

recursively sorted using the same `QSORT` function. The closure is exploited to define comparison functions.

## 4.14 A Monad Is Just a Monoid in the Category of Endofunctors<sup>†</sup>, What's the Problem?

And obviously it's time to talk about the Monad. What is it? Well, I don't know about you but the section title is surely not very helpful...

A monad can be seen as a container that holds whatever the value it can accept, and allows alteration of the value by *binding*, and its internals can be evaluated later.

...Pretty vague, eh? But thanks to its broad definition, it can be used to implement many things. For example, let me show you how monad can be used to add memoisation (and thus making it faster!) to the aforementioned Fibonacci sequence generator, without clobbering a global variable, of course:

---

```
10 FIB = [N,M] ~> IF (M==UNDEFINED) THEN FIB(N, MRET({1, 1})) ELSE IF
    (LEN(MJOIN(M)) >= N) THEN HEAD(MJOIN(M)) ELSE FIB(N, M >>= ([XS]
    ~> MRET((XS(0) + XS(1)) ! XS)))
20 FOR K = 1 TO 10
30 PRINT FIB(K); " ";
40 NEXT
```

---

In Line 10, `>>=` (a *bind* operator) extracts inner value of the monad `M` as `XS`<sup>‡</sup> (which is an array), inserts `XS(0) + XS(1)` before the `XS` (the `!` operator is doing the task), then wraps the new array into a monad using `MRET` function and then passes the new monad into the `FIB`'s recursive call; if array length of the inner value of the monad reaches desired length, returns head-element of the value.

Thanks to the Monad holding the results from previous runs, making double-recursion run unnecessary unlike the previous attempts, the Fibonacci sequence generator now runs much faster.

And this is exactly what *memoisation* is, remembering (or *memoing*) the previous results.

---

<sup>†</sup>Saunders Mac Lane, *Categories for the Working Mathematician*

<sup>‡</sup>Stands for *Xs* (plural form of *X*)

# Chapter 5

## Language Reference

This chapter describes the Terran BASIC language.

### 5.1 Metasyntax

In the descriptions of BASIC syntax, these conventions apply.

- **VERBATIM** — Type exactly as shown
- **IDENTIFIER** — Replace *identifier* with appropriate metavariable
- **[a]** — Words within square brackets are optional
- **{a|b}** — Choose either **a** or **b**
- **[a|b]** — Optional version of the above
- **a...** — The preceding entity can be repeated

### 5.2 Definitions

A *Program Line* consists of a line number followed by a *Statements*. Program Lines are terminated by a line break or by the end-of-the-file.

A *Line Number* is an integer within the range of  $[0..2^{53} - 1]$ .

A *Statement* is a special form of code which has special meaning. A program line can be composed of 1 or more statements, separated by colons. For the details of statements available in Terran BASIC, see 7.

```
STATEMENT [ : STATEMENT ] ...
```

An *Expression* is rather normal program lines, e.g. mathematical equations and function calls. The expression takes one of the following forms. For the details of functions available in Terran BASIC, see 8.

```
VARIABLE_OR_FUNCTION
```

```
( EXPRESSION )
```

```
IF EXPRESSION THEN EXPRESSION [ELSE EXPRESSION]
```

```
FUNCTION ( [EXPRESSION { , | ; } [{ , | ; }]] )
```

```
FUNCTION [EXPRESSION { , | ; } [{ , | ; }]]
```

```
EXPRESSION BINARY_OPERATOR EXPRESSION
```

```
UNARY_OPERATOR EXPRESSION
```

An *Array* takes following form:

```
ARRAY_NAME ( EXPRESSION [ , EXPRESSION ] ... )
```

## 5.3 Literals

### 5.3.1 String Literals

String literals take the following form:

```
" [CHARACTERS] "
```

where `CHARACTERS` is a 1- or more repetition of ASCII-printable letters.\*

To print out graphical letters outside of ASCII-printable, use string concatenation with `CHR` function, or use `EMIT` function.

### 5.3.2 Numeric Literals

Numeric literals take one of the following forms:

```
[+|-] [0|1|2|3|4|5|6|7|8|9]... [.] [0|1|2|3|4|5|6|7|8|9]...
```

```
0{x|X} [0|1|2|3|4|5|6|7|8|9]...
```

```
0{b|B} [0|1|2|3|4|5|6|7|8|9]...
```

Hexadecimal and binary literals are always interpreted as *unsigned* integers. They must range between  $[0..2^{53} - 1]$ .

### 5.3.3 Variables

Variable names must not begin with a figure and all characters of the name must be letters `A-Z`, figures `0-9` and underscores `_`. Variable names must not be identical to reserved words, but may *contain* one. Variable names are case-insensitive.

---

\*In other words, `0x20..0x7E`

Unlike conventional BASIC dialects (especially GW-BASIC), name pool of variables is shared between all the types. For example, if you have a numeric variable `A`, and define an array named `A` later in the program, the new array will overwrite your numeric `A`.

Furthermore, *sigils* are not used in the Terran BASIC and attempting to use one will raise syntax-error (the `$` is an operator in Terran BASIC) or undefined behaviours.

5.3.4 Types

Types of data recognised by Terran BASIC are distinguished by some arcane magic of Javascript auto-casing mumbo-jumbo

Type	Range	Precision
String	As many as the machine can handle	always precise
Integer	$\pm 2^{53} - 1$	exact within the range
Float	$\pm 4.9406564584124654 \times 10^{-324} - \pm 1.7976931348623157 \times 10^{308}$	about 16 significant figures

5.4 Operators

5.4.1 Order of Precedence

The order of precedence of the operators is as shown below, lower numbers mean they have higher precedence (more tightly bound)

Order	Op	Associativity	Order	Op	Associativity
1	<code>\</code>	Right	14	<code>AND</code>	Left
2	<code>^</code>	Right	15	<code>OR</code>	Left
3	<code>*</code> <code>/</code> <code>\</code>	Left	16	<code>TO STEP</code>	Left
4	<code>MOD</code>	Left	17	<code>!</code>	Right
5	<code>+</code> <code>-</code>	Left	18	<code>~</code>	Left
6	<code>NOT</code> <code>BNOT</code>	Left	19	<code>#</code>	Left
7	<code>&lt;&lt;</code> <code>&gt;&gt;</code>	Left	20	<code>.</code> <code>\$</code>	Right
8	<code>&lt;</code> <code>=&lt;</code> <code>&lt;=</code>	Left	21	<code>&amp;</code>	Left
	<code>&gt;</code> <code>=&gt;</code> <code>&gt;=</code>		22	<code>~&lt;</code>	Left
9	<code>==</code> <code>&lt;&gt;</code> <code>&gt;&lt;</code>	Left	23	<code>&lt;\$&gt;</code>	Right
10	<code>MIN</code> <code>MAX</code>	Left	24	<code>&lt;*&gt;</code> <code>&lt;~&gt;</code>	Left
11	<code>BAND</code>	Left	25	<code>@</code>	Right
12	<code>BXOR</code>	Left	26	<code>~&gt;</code> <code>&gt;&gt;~</code> <code>&gt;&gt;=</code>	Right
13	<code>BOR</code>	Left	27	<code>=</code>	Right

## Examples

- Exponentiation is more tightly bound than negation:  $-1^2 == -(1^2) == -1$  but  $(-1)^2 == 1$
- Exponentiation is right-associative:  $4^3^2 == 4^{(3^2)} == 262144$ . This behaviour is *different* from GW-BASIC in which its exponentiation is left-associative.

### 5.4.2 Mathematical Operators

Mathematical operators operate on expressions that return numeric value only, except for the `+` operator which will take the action of string concatenation if either of the operands is non-numeric.

Code	Operation	Result
$x = y$	Assignment	Assigns $y$ into $x$
$x \wedge y$	Exponentiation	$x$ raised to the $y$ th power
$x * y$	Multiplication	Product of $x$ and $y$
$x / y$	Division	Quotient of $x$ and $y$
$x \setminus y$	Truncated Division	Integer quotient of $x$ and $y$
$x \text{ MOD } y$	Modulo	Integer remainder of $x$ and $y$ with sign of $x$
$x + y$	Addition	Sum of $x$ and $y$
$x - y$	Subtraction	Difference of $x$ and $y$
$+ x$	Unary Plus	Value of $x$
$- x$	Unary Minus	Negative value of $x$
$x \text{ MIN } y$	Minimum	Lesser value of two
$x \text{ MAX } y$	Maximum	Greater value of two

## Notes

- Type conversion rule follows underlying Javascript implementation. In other words, *only the god knows*.
- The expression  $0^0$  will return `1`, even though the expression is indeterminant.

## Errors

- Any expression that results `NaN` or `Infinity` in Javascript will return some kind of errors, mainly `Division by zero`.
- If  $x < 0$  and  $y$  is not integer,  $x^y$  will raise `Illegal function call`.

### 5.4.3 Comparison Operators

Comparison operator can operate on numeric and string operands. String operands will be automatically converted to numeric value if they can be; if one operand is numeric and other is a non-numeric string, the former will be converted to a string value.



Code	Operation	Result
<code>x == y</code>	Equal	True if <code>x</code> equals <code>y</code>
<code>x &lt;&gt; y</code> <code>x &lt;&gt; y</code>	Not equal	False if <code>x</code> equals <code>y</code>
<code>x &lt; y</code>	Less than	True if <code>x</code> is less than <code>y</code>
<code>x &gt; y</code>	Greater than	True if <code>x</code> is greater than <code>y</code>
<code>x &lt;= y</code> <code>x &lt;= y</code>	Less than or equal	False if <code>x</code> is greater than <code>y</code>
<code>x &gt;= y</code> <code>x &gt;= y</code>	Greater than or equal	False if <code>x</code> is less than <code>y</code>

When comparing strings, the ordering is as follows:

- Two strings are equal only when they are of the same length and every code-point of the first string is identical to that of the second. This includes any whitespace or unprintable characters.
- Each character position of the string is compared starting from the leftmost character. When a pair of different characters is encountered, the string with the character of lesser codepoint is less than the string with the character of greater codepoint.
- If the strings are of different length, but equal up to the length of the shorter string, then the shorter string is less than the longer string.

### 5.4.4 Bitwise Operators

Bitwise operators operate on unsigned integers only. Floating points are truncated<sup>†</sup> to integers.

Code	Operation	Result
<code>x &lt;&lt; y</code>	Bitwise Shift Left	Shifts entire bits of <code>x</code> by <code>y</code>
<code>x &gt;&gt; y</code>	Bitwise Shift Right	Shift entire bits <code>x</code> by <code>y</code> , including sign bit
<code>BNOT x</code>	Ones' complement	$-x - 1$
<code>x BAND y</code>	Bitwise conjunction	Bitwise AND of <code>x</code> and <code>y</code>
<code>x BOR y</code>	Bitwise disjunction	Bitwise OR of <code>x</code> and <code>y</code>
<code>x BXOR y</code>	Bitwise add-with-no-carry	Bitwise XOR of <code>x</code> and <code>y</code>

### 5.4.5 Boolean Operators

Boolean operators operate on boolean values. If one of the operands is not boolean, it will be cast to an appropriate boolean value. See 3.1 for casting rules.

Code	Operation	Result
<code>NOT x</code>	Logical negation	True if <code>x</code> is false and vice versa
<code>x AND y</code>	Bitwise conjunction	True if <code>x</code> and <code>y</code> are both true
<code>x OR y</code>	Bitwise disjunction	True if <code>x</code> or <code>y</code> is true, or both are true

<sup>†</sup>Truncated towards zero.

### 5.4.6 Generator Operators

Generator operators operate on numeric values and generators to create and modify a generator.

Code	Result
$x \text{ TO } y$	Creates an generator that counts from $x$ to $y$
$x \text{ STEP } y$	Modifies a counting stride of the generator $x$ into $y$

### 5.4.7 Array Operators

Array operators operate on arrays and numeric values.

Code	Operation	Result
$x ! y$	Cons	Prepends a value of $x$ into an array of $y$
$x \sim y$	Push	Appends a value of $y$ into an array of $x$
$x \# y$	Concat	Concatenates two arrays

Arbitrary arrays can be constructed using empty-array constant **NIL**.

### 5.4.8 Function Operators

Function operators operate on a function and some values.

Code	Operation	Result
$f \sim x$	Curry	Partially apply $x$ into the first parameter of the function $f$ and returns the resulting function
$f \$ x$	Apply	Evaluates single-parameter function $f(x)$ and returns the value
$x \& f$	Pipe	Reversed version of the $(\$)$
$f . g$	Compo	Creates a new function $f \circ g$ where function $g$ is pipelined into the function $f$
$[x, y, \dots] \rightsquigarrow e$	Closure	Creates a closure (anonymous function) from one or more parameters $x, y, \dots$ and an expression $e$

*Currying* is an operation which returns new function that has given value applied to the original function's first parameter. See 4.12 for tutorials.

*Applying* is, as the name suggests, applies the right-hand value onto the function on the left-hand, and returns the result of that operation.

*Pipe* “pipes” the left-hand value into the right-hand function, effectively working as the reversed *Apply*.

Function *Composition* is an operation which pipelines the result from one function into the input of another function, creating entirely a new function.

*Closure* defines anonymous function with given parameters and the expression, and returns the function.

### 5.4.9 Functor Operators

Functor operators operate on one or more functions and list of values (or, *functors*).

Code	Operation	Result
<code>f &lt;\$&gt; xs</code>	Map	Executes built-in MAP function on a function <i>f</i> and the list of values <i>xs</i>
<code>fs &lt;~&gt; xs</code>	Curry Map	Curries <i>xs</i> to the each function in a list of function <i>fs</i> , and concatenates the results
<code>fs &lt;*&gt; xs</code>	Ap	MAPs the list of values <i>xs</i> for each function in a list of functions <i>fs</i> , and concatenates the results

Curry Map can be used to map two or more arguments into a function, take this example code:

---

```
1 STRINGIFY=[X,Y]~>X + " x " + Y + " = " + (X*Y)
10 K=STRINGIFY <~> 2 TO 9 <*> 1 TO 9
20 FOREACH S IN K : PRINT K : NEXT
```

---

`STRINGIFY` will print a row in the multiplication table and `<~>` and `<*>` is used to map `2 TO 9` and `1 TO 9` into the `STRINGIFY`. Do not mistakenly assume that, in this code, `<*>` magically combines two generators and pass them to the function; it's the `<~>` that gets evaluated first.

But if you want to map three- or more, the syntax gets less obvious:

---

```
1 CAT3=[X,Y,Z]~>X + "/" + Y + "/" + Z
10 PRINT(CAT3 <~> 1 TO 3 <~> 4 TO 6 <*> 7 TO 9)
```

---

Note that `<~>` is used in between `1 TO 3` and `4 TO 6` because we're *currying*.

#### Notes

- Curry Map may receive just a function for its left-hand value.

### 5.4.10 Monad Operators

Monad operators operate on monads and some values.

Code	Operation	Result
$m\ a \gg= f$	Bind	Sends inner value $a$ from the monad $m\ a$ into the function $f$ that returns new monad $m\ b$ using the $a$
$m\ a \gg\sim m\ b$	Sequence	Discards inner value of the monad $m\ a$ and returns new monad $m\ b$
$\backslash m$	Join	Shorthand for <code>MJOIN (m)</code>
$@m$	Return	Shorthand for <code>MRET (m)</code>

## Notes

- Not all monad types obey monad laws, especially the funseq-monad.

## 5.5 Constants

Some variables are pre-defined on the language itself and cannot be modified; such variables are called *constants*.

Name	Type	Value	Description
PI	Number	3.141592653589793	$\pi$
TAU	Number	6.283185307179586	$2\pi$
EULER	Number	2.718281828459045	Euler's number $e$
NIL	Array	Empty Array	Used to construct arbitrary array using CONS-operator
ID	Function	<code>[X] ~&gt; X</code>	An identity function
UNDEFINED	Undefined	undefined	Undefined

## 5.6 Syntax In EBNF

If you're *that* into the language theory of computer science, texts above are just waste of bytes/inks/pixel-spaces/whatever; this little section should be more than enough!

```
(* quick reference to EBNF *)
(* { word } = word is repeated 0 or more times *)
(* [ word ] = word is optional (repeated 0 or 1 times) *)

line =
  | linenumbr , stmt , {":" , stmt}
  | linenumbr , "REM" , ? basically anything ? ;
linenumbr = digits ;

stmt =
  "REM" , ? anything ?
  | "IF" , expr_sans_asgn , "THEN" , stmt , ["ELSE" , stmt]
  | "DEFUN" , [ident] , "(" , [ident , {" , " , ident}] , ")" , "=" , expr
  | "ON" , expr_sans_asgn , ("GOTO" | "GOSUB" ) , expr_sans_asgn , {" , " , expr_sans_asgn}
  | "(" , stmt , ")"
  | expr ; (* if the statement is 'lit' and contains only one word, treat it as function_call
           e.g. NEXT for FOR loop *)
```

```

expr = (* this basically blocks some funny attempts such as using DEFUN as anon function
        because everything is global in BASIC *)
    ? empty string ?
    | lit
    | "{" , [expr , {" , " , expr}] , "}"
    | "(" , expr , ")"
    | ident_tuple
    | "IF" , expr_sans_asgn , "THEN" , expr , ["ELSE" , expr]
    | ("FOR"|"FOREACH") , expr
    | expr , op , expr
    | op_uni , expr
    | kywd , expr - "("
    | function_call ;

expr_sans_asgn = ? identical to expr except errors out whenever "=" is found ? ;

ident_tuple = "[" , ident , {" , " , ident} , "]" ;

function_call =
    ident , "(" , [expr , {argsep , expr} , [argsep]] , ")"
    | ident , expr , {argsep , expr} , [argsep] ;
kywd = ? words that exists on the list of predefined function that are not operators ? ;

(* don't bother looking at these, because you already know the stuff *)

argsep = "," | ";" ;
ident = alph , [digits] ; (* variable and function names *)
lit = alph , [digits] | num | string ; (* ident + numbers and string literals *)
op = "~" | "*" | "/" | "\" | "MOD" | "+" | "-" | "<<" | ">>" | "<" | ">"
    | "<=" | "<" | ">=" | ">" | "==" | "<" | ">" | "MIN" | "MAX" | "BAND" | "BXOR" | "BOR"
    | "AND" | "OR" | "TO" | "STEP" | "!" | "~" | "#" | "." | "$" | "&" | "-<" | "<$>" | "<*>"
    | "<->" | "~>" | ">>~" | ">>=" | "=" ;
op_uni = "-" | "+" | "NOT" | "BNOT" | "~" | "@" ;

alph = letter | letter , alph ;
digits = digit | digit , digits ;
hexdigits = hexdigit | hexdigit , hexdigits ;
bindigits = bindigit | bindigit , bindigits ;
num = digits | digits , "." , [digits] | "." , digits
    | ("0x"|"0X") , hexdigits
    | ("0b"|"0B") , bindigits ; (* sorry, no e-notation! *)
visible = ? ASCII 0x20 to 0x7E ? ;
string = "'" , {visible} , "'" ;

letter = "A" | "B" | "C" | "D" | "E" | "F" | "G" | "H" | "I" | "J" | "K" | "L" | "M" | "N"
    | "O" | "P" | "Q" | "R" | "S" | "T" | "U" | "V" | "W" | "X" | "Y" | "Z" | "a" | "b"
    | "c" | "d" | "e" | "f" | "g" | "h" | "i" | "j" | "k" | "l" | "m" | "n" | "o" | "p"
    | "q" | "r" | "s" | "t" | "u" | "v" | "w" | "x" | "y" | "z" | "_" ;
digit = "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9" ;
hexdigit = "A" | "B" | "C" | "D" | "E" | "F" | "a" | "b" | "c" | "d" | "e" | "f" | "0" | "1"
    | "2" | "3" | "4" | "5" | "6" | "7" | "8" | "9" ;
bindigit = "0" | "1" ;

(* all possible token states: lit num op bool qot paren sep *)
(* below are schematic of trees generated after parsing the statements *)

IF (type: function, value: IF)
1. cond
2. true
[3. false]

```

FOR (type: function, value: FOR)

1. expr (normally (=) but not necessarily)

DEFUN (type: function, value: DEFUN)

1. funcname (type: lit)
  1. arg0 (type: lit)
  - [2. arg1]
  - [3. argN...]
2. stmt

ON (type: function, value: ON)

1. testvalue
2. functionname (type: lit)
3. arg0
- [4. arg1]
- [5. argN...]

FUNCTION\_CALL (type: function, value: PRINT or something)

1. arg0
2. arg1
- [3. argN...]

LAMBDA (type: op, value: "->")

1. undefined (type: closure\_args, value: undefined)
  1. arg0 (type: lit)
  - [2. arg1]
  - [3. argN...]
2. stmt

ARRAY\_CONSTRUCTOR (type: function, value: undefined)

1. 0th element of the array
2. 1st element of the array
- [3. Nth element of the array...]

# Chapter 6

## Commands

This chapter describes commands accepted by the Terran BASIC editor.

### 6.1 The Editor

When you first launch the Terran BASIC, all you can see is some generic welcome text and two letters: `OK`. Sure, you can just start type away your programs and type `run` to execute them, there are more things you can do with.

#### 6.1.1 CATALOG

```
CATALOG
```

Shows the directory contents of the working directory.

#### 6.1.2 CLS

```
CLS
```

Clears text view and moves text cursor to top-left.

#### 6.1.3 DELETE

```
DELETE LINE
```

```
DELETE LINE_START LINE_END
```

Deletes a given line. If two arguments were given, deletes any lines between them, start- and end-inclusive.

#### 6.1.4 LOAD

```
LOAD FILENAME
```

Loads BASIC program by the file name. Default working directory for Terran BASIC is `/home/basic.*`

---

\*This is a directory within the emulated disk. On the host machine, this directory is typically `PWD/assets/diskN/home/basic`, where `PWD` is working directory for the TSVM, `diskN` is a number of the disk.

### 6.1.5 LIST

```
LIST [LINE_NUMBER]
```

```
LIST [LINE_FROM LINE_TO]
```

Displays a BASIC program that currently has been typed. When no arguments were given, shows the entire program; when single line number was given, displays that line; when a range of line numbers was given, displays those lines.

### 6.1.6 NEW

```
NEW
```

Immediately deletes the program that currently has been typed and resets the environment: wipes out `DATA`, variables and line labels, and resets `DATA` cursor and `OPTIONBASE` to zero

### 6.1.7 RENUM

```
RENUM
```

Re-numbers program line starting from 10 and incrementing by 10s. Jump targets will be re-numbered accordingly. Nonexisting jump targets will be replaced with `undefined`.<sup>†</sup>

### 6.1.8 RUN

```
RUN
```

Executes BASIC program that currently has been typed. Execution can be arbitrarily terminated with Ctrl-C key combination (except in `INPUT` mode).

### 6.1.9 SAVE

```
SAVE FILENAME
```

Saves BASIC program that currently has been typed. Existing files are overwritten *silently*.

### 6.1.10 SYSTEM

```
SYSTEM
```

---

<sup>†</sup>This behaviour is simply Javascript's null-value leaking into the BASIC. This is nonstandard behaviour and other Terran BASIC implementations may act differently.



Exits Terran BASIC.

# Chapter 7

## Statements

A Program line is composed of a line number and one or more statements. Multiple statements are separated by colons `:`.

### 7.1 IF

```
IF TRUTH_VALUE THEN TRUE_EXPRESSION [ELSE FALSE_EXPERSSION]
```

If `TRUTH_VALUE` is truthy, executes `TRUE_EXPRESSION`. If `TRUTH_VALUE` is falsy and `FALSE_EXPERSSION` is specified, executes that expression; otherwise, the next line or next statement will be executed.

#### Notes

- **IF** is both statement and expression. You can use IF-clause after **ELSE**, or within functions as well, for example.
- **THEN** is *not* optional, this behaviour is different from most of the BASIC dialects.
- Also unlike the most dialects, **GOTO** cannot be omitted; doing so will make the number be returned to its parent expression.

### 7.2 ON

```
ON INDEX_EXPRESSION {GOTO|GOSUB} LINE0 [, LINE1]...
```

Jumps to the line number returned by the `INDEX_EXPRESSION`. If the result is outside of the range of the arguments, no jump will be performed.

#### Parameters

- `LINEn` can be a number, numeric expression (aka equations) or a line label.
- When `OPTIONBASE 1` is used within the program, `LINEn` starts from 1 instead of 0.

## 7.3 DEFUN

*There it is, the DEFUN. All those new-fangled parser\* and paradigms† are tied to this very statement on Terran BASIC, and only Wally knows its secrets...*

```
DEFUN NAME ( [ARGS0 [, ARGS1]...] ) = EXPRESSION
```

With the aid of other statements‡ and functions, DEFUN will allow you to ascend from traditional BASIC and do godly things such as *recursion*§ and *functional programming*.

Oh, and you can define your own function, in traditional DEF FN sense.

### Parameters

- `NAME` must be a valid variable name.
- `ARGSn` must be valid variable names, but can be a name of variables already used within the BASIC program; their value will not be affected nor be used.

---

\*A computer program that translates program code entered by you into some data bits that only it can understand.

†A guidance to in which way you must think to assimilate your brain into the computer-overlord.

‡Actually, only the IF is useful. Use closure expression for more sophisticated function definition.

§See recursion.

# Chapter 8

## Functions

Functions are a form of expression that may take input arguments surrounded by parentheses. Most of the traditional BASIC *statements* that does not return a value are *functions* in Terran BASIC, and like those, while Terran BASIC functions can be called without parentheses, it is highly *discouraged* because of the ambiguities in syntax. **Always use parentheses on function call!**

### 8.1 Mathematical

#### 8.1.1 ABS

```
Y = ABS (X)
```

Returns absolute value of `X`.

#### 8.1.2 ACO

```
Y = ACO (X)
```

Returns inverse cosine of `X`.

#### 8.1.3 ASN

```
Y = ASN (X)
```

Returns inverse sine of `X`.

#### 8.1.4 ATN

```
Y = ATN (X)
```

Returns inverse tangent of `X`.

#### 8.1.5 CBR

```
Y = CBR (X)
```

Returns cubic root of `X`.

### 8.1.6 CEIL

```
Y = CEIL(X)
```

Returns integer value of `X`, truncated towards positive infinity.

### 8.1.7 COS

```
Y = COS(X)
```

Returns cosine of `X`.

### 8.1.8 COSH

```
Y = COSH(X)
```

Returns hyperbolic cosine of `X`.

### 8.1.9 EXP

```
Y = EXP(X)
```

Returns exponential of `X`, i.e.  $e^X$ .

### 8.1.10 FIX

```
Y = FIX(X)
```

Returns integer value of `X`, truncated towards zero.

### 8.1.11 FLOOR, INT

```
Y = FLOOR(X)
```

```
Y = INT(X)
```

Returns integer value of `X`, truncated towards negative infinity.

### 8.1.12 LEN

```
Y = LEN(X)
```

Returns length of `X`. `X` can be either a string or an array.

### 8.1.13 LOG

$$Y = \text{LOG}(X)$$

Returns natural logarithm of  $X$ .

### 8.1.14 ROUND

$$Y = \text{ROUND}(X)$$

Returns closest integer value of  $X$ , rounding towards positive infinity.

### 8.1.15 RND

$$Y = \text{RND}(X)$$

Returns a random number within the range of  $[0..1)$ . If  $X$  is zero, previous random number will be returned; otherwise new random number will be returned.

### 8.1.16 SIN

$$Y = \text{SIN}(X)$$

Returns sine of  $X$ .

### 8.1.17 SINH

$$Y = \text{SINH}(X)$$

Returns hyperbolic sine of  $X$ .

### 8.1.18 SGN

$$Y = \text{SGN}(X)$$

Returns sign of  $X$ : 1 for positive, -1 for negative, 0 otherwise.

### 8.1.19 SQR

$$Y = \text{SQR}(X)$$

Returns square root of  $X$ .

### 8.1.20 TAN

```
Y = TAN(X)
```

Returns tangent of `X`.

### 8.1.21 TANH

```
Y = TANH(X)
```

Returns hyperbolic tangent of `X`.

### 8.1.22 TEST

```
Y = TEST(X)
```

Tests if the value is truthy or not. If the value must be interpreted as `TRUE` for the if-statement, said value is *truthy*.

## 8.2 Input

### 8.2.1 CIN

```
S = CIN()
```

Waits for the user input and returns it.

### 8.2.2 DATA

```
DATA CONST0 [, CONST1]...
```

Adds data that can be read by `READ` function. `DATA` declarations need not be reachable in the program flow.

### 8.2.3 DGET

```
S = DGET()
```

Fetches a data declared from `DATA` statements and returns it, incrementing the `DATA` position.

### 8.2.4 DIM

```
Y = DIM(X)
```

Returns array with size of `X`, all filled with zero.

### 8.2.5 GETKEYSDOWN

```
K = GETKEYSDOWN()
```

Stores array that contains keycode of keys held down into the given variable.

Actual keycode and the array length depends on the machine: in TSVM, array length will be fixed to 8. For the list of available keycodes, see 10.

### 8.2.6 INPUT

```
INPUT VARIABLE
```

Prints out `?` to the console and waits for user input. Input can be any length and terminated with the return key. The input will be stored in the given variable.

This behaviour is to keep compatibility with the traditional BASIC. For function-like usage, use `CIN` instead.

### 8.2.7 READ

```
READ VARIABLE
```

Assigns data declared from `DATA` statements to given variable. Reading starts at the current `DATA` position, and the data position will be incremented by one. The position is reset to the zero by the `RUN` command.

This behaviour is to keep the compatibility with the traditional BASIC. For function-like usage, use `DGET` instead.

## 8.3 Output

### 8.3.1 EMIT

```
EMIT( EXPR [{,|;} EXPR]... )
```

Prints out characters corresponding to given number on the code page being used. For the code page itself, see 10.2.

`EXPR` is numeric expression.



### 8.3.2 PRINT

```
PRINT ( EXPR [{, | ;} EXPR]... )
```

Prints out given string expressions.

**EXPR** is a string, numeric expression, or array.

**PRINT** is one of the few function that differentiates two style of argument separator: **;** will simply concatenate two expressions (unlike traditional BASIC, numbers will not have surrounding spaces), **,** tabulates the expressions.

## 8.4 Program Manipulation

### 8.4.1 CLEAR

```
CLEAR
```

Clears all declared variables.

### 8.4.2 END

```
END
```

Stops program execution and returns control to the user.

### 8.4.3 FOR

```
FOR LOOPVAR = START TO STOP [STEP STEP]
```

```
FOR LOOPVAR = GENERATOR
```

Starts a FOR–NEXT loop.

Initially, **LOOPVAR** is set to **START** then statements between the **FOR** statement and corresponding **NEXT** statements are executed and **LOOPVAR** is incremented by **STEP**, or by 1 if **STEP** is not specified. The program flow will continue to loop around until **LOOPVAR** is outside the range of **START–STOP**. The value of the **LOOPVAR** is equal to **STOP + STEP** when the looping finishes.

### 8.4.4 FOREACH

```
FOREACH LOOPVAR IN ARRAY
```

Same as **FOR** but fetches **LOOPVAR** from given **ARRAY**.

### 8.4.5 GOSUB

```
GOSUB LINENUM
```

Jumps to a subroutine at `LINENUM`. The next `RETURN` statements makes program flow to jump back to the statement after the `GOSUB`.

`LINENUM` can be either a numeric expression or a Label.

### 8.4.6 GOTO

```
GOTO LINENUM
```

Jumps to `LINENUM`.

`LINENUM` can be either a numeric expression or a Label.

### 8.4.7 LABEL

```
LABEL NAME
```

Puts a name onto the line number where the statement is located.

#### Notes

- `NAME` must be a valid variable name.

### 8.4.8 NEXT

```
NEXT
```

Iterates FOR–NEXT loop and increments the loop variable from the most recent `FOR` statement and jumps to that statement.

### 8.4.9 RESTORE

```
RESTORE
```

Resets the `DATA` pointer.

### 8.4.10 RETURN

```
RETURN
```

Returns from the `GOSUB` statement.

## 8.5 String Manipulation

### 8.5.1 CHR

```
CHAR = CHR(X)
```

Returns the character with code point of `X`. Code point is a numeric expression in the range of `[0 – 255]`.

### 8.5.2 LEFT

```
SUBSTR = LEFT( STR , NUM_CHARS )
```

Returns the leftmost `NUM_CHARS` characters of `STR`.

### 8.5.3 MID

```
SUBSTR = MID( STR , POSITION , LENGTH )
```

Returns a substring of `STR` starting at `POSITION` with specified `LENGTH`.

When `OPTIONBASE 1` is specified, the position starts from 1; otherwise it will start from 0.

### 8.5.4 RIGHT

```
SUBSTR = RIGHT( STR , NUM_CHARS )
```

Returns the rightmost `NUM_CHARS` characters of `STR`.

### 8.5.5 SPC

```
STR = SPC( STR , NUM_CHARS )
```

Returns a string of `NUM_CHARS` spaces.

## 8.6 Array Manipulation

### 8.6.1 HEAD

```
K = HEAD(X)
```

Returns the head element of the array `X`.

### 8.6.2 INIT

```
K = INIT(X)
```

Constructs the new array from array `X` that has its last element removed.

### 8.6.3 LAST

```
K = LAST(X)
```

Returns the last element of the array `X`.

### 8.6.4 TAIL

```
K = TAIL(X)
```

Constructs the new array from array `X` that has its head element removed.

## 8.7 Monad Manipulation

### 8.7.1 MJOIN

```
K = MJOIN(M)
```

Returns the inner value of the given monad.

### 8.7.2 MRET

```
M = MRET(X)
```

Returns a value-monad that contains a given value.

## 8.8 Graphics

### 8.8.1 CLPX

```
CLPX
```

Clears plotted pixels.

### 8.8.2 CLS

```
CLS
```

Clears text view and moves text cursor to top-left.

### 8.8.3 GOTOYX

```
GOTOYX( ROW , COLUMN )
```

Moves text cursor to given row and column.

When `OPTIONBASE 1` is specified, first row and column will be 1, otherwise it will be 0.

### 8.8.4 PLOT

```
PLOT( X_POS , Y_POS , COLOUR )
```

Plots a pixel to the framebuffer of the display, at XY-position of `X_POS` and `Y_POS`, with colour of `COLOUR`. For the available colours, see 10.3

Top-left corner of the pixel will be 1 if `OPTIONBASE 1` is specified, otherwise it will be 0.

## 8.9 Meta

### 8.9.1 OPTIONBASE

```
OPTIONBASE { 0 | 1 }
```

Specifies at which number the array/string/pixel indices begin.

### 8.9.2 OPTIONDEBUG

```
OPTIONDEBUG { 0 | 1 }
```

Specifies whether or not the debugging messages should be printed out. The messages will be printed out to the *serial debugging console*, or to the stdout.

**Big Warning Sign:** Do not turn debug mode on unless you know what you're doing; debug mode will severely slow down the interpreter and literally gigabytes of log messages will pile up in a minute or two of the execution.

### 8.9.3 OPTIONTRACE

```
OPTIONTRACE { 0 | 1 }
```

Specifies whether or not the line numbers should be printed out. The messages will be printed out to the *serial debugging console*, or to the stdout.

### 8.9.4 TYPEOF

```
X = typeof( VALUE )
```

Returns a type of given value.

BASIC Type	Returned Value	BASIC Type	Returned Value
Number	num	Generator	generator
Boolean	bool	User Function	usrdefun
String	str	Monad	<subtype>-monad
Array	array		

## 8.10 System

### 8.10.1 PEEK

```
BYTE = peek( MEM_ADDR )
```

Returns whatever the value stored in the `MEM_ADDR` of the Scratchpad- or Machine Memory.

Address mirroring, illegal access, etc. are entirely up to the virtual machine which the BASIC interpreter is running on.

For Machine Memory addresses, see 10.4.

### 8.10.2 POKE

```
poke( MEM_ADDR , BYTE )
```

Puts a `BYTE` into the `MEM_ADDR` of the Scratchpad- or Machine Memory.

## 8.11 Higher-order Function

### 8.11.1 DO

```
do( EXPR0 [ ; EXPR1 ]... )
```

Executes `EXPRn`s sequentially.

### 8.11.2 FILTER

```
NEWLIST = FILTER( FUNCTION , FUNCTOR )
```

Returns an array of values from the `FUNCTOR` that passes the given function. i.e. values that makes `FUNCTION(VALUE_FROM_FUNCTOR)` true.

#### Parameters

- `FUNCTION` is a user-defined or builtin function with single parameter.
- `FUNCTOR` is either an array or a generator.

### 8.11.3 FOLD

```
NEWVALUE = FOLD( FUNCTION , INIT_VALUE , FUNCTOR )
```

Iteratively applies given function with accumulator and the value from the `FUNCTOR`, returning the final accumulator. Accumulator will be set to `INIT_VALUE` before iterating over the functor. In the first execution, the accumulator will be set to `ACC=FUNCTION(ACC, FUNCTOR(0))`, and the execution will continue to remaining values within the functor until all values are consumed. The `FUNCTOR` will not be modified after the execution.

#### Parameters

- `FUNCTION` is a user-defined function with two parameters: first parameter being accumulator and second being a value.

### 8.11.4 MAP

```
NEWLIST = MAP( FUNCTION , FUNCTOR )
```

Applies given function onto the every element in the functor, and returns an array that contains such items. i.e. returns transformation of `FUNCTOR` of which the transformation is `FUNCTION`. The `FUNCTOR` will not be modified after the execution.

#### Parameters

- `FUNCTION` is a user-defined or builtin function with single parameter.

## **Part II**

# **Implementation**



# Chapter 9

## Interpreter

This chapter documents the reference implementation of Terran BASIC.

### 9.1 Resolving Variables

When a variable is `resolved`, an object with instance of `BasicVar` is returned. A `bvType` of Javascript value is determined using `JStoBASICtype`.

Typical User Input	TYPEOF(Q)	Instanceof
<code>Q=42.195</code>	<code>num</code>	<i>primitive</i>
<code>Q=42&gt;21</code>	<code>boolean</code>	<i>primitive</i>
<code>Q="BASIC!"</code>	<code>string</code>	<i>primitive</i>
<code>Q=DIM(12)</code>	<code>array</code>	<code>Array (JS)</code>
<code>Q=1 TO 9 STEP 2</code>	<code>generator</code>	<code>ForGen</code>
<code>DEFUN Q(X)=X+3</code>	<code>usrdefun</code>	<code>BasicAST</code>
<code>Q=MRET(X)</code>	<code>value-monad</code>	<code>BasicMemoMonad</code>
<code>Q=F.G</code>	<code>funseq-monad</code>	<code>BasicFunSeqMonad</code>
<code>Q=UNDEFINED</code>	<code>undefined</code>	<i>primitive</i>

#### Notes

- For non-monadic value of Q, `TYPEOF(Q)` is identical to the variable's `bvType`: the function simply returns `BasicVar.bvType`; for monadic value, `TYPEOF(Q)` will be *subtype-monad*, with *subtype* corresponds to `monadObject.mType`.
- Funseq-monad is a pseudo-monad: they do not obey monad laws.
- Do note that all resolved variables have `troType` of `Lit`, see next section for more information.

### 9.2 Unresolved Values

Unresolved variables has JS-object of `troType`, with *instanceof* `SyntaxTreeReturnObj`. Its properties are defined as follows:

Properties	Description
<code>troType</code>	Type of the TRO (Tree Return Object)
<code>troValue</code>	Value of the TRO
<code>troNextLine</code>	Pointer to next instruction, array of: <code>[#line, #statement]</code>

Following table shows which BASIC object can have which `troType`:

BASIC Type	troType
Any Variable	lit
Boolean	bool
Number	num
String	string
Array	array
Generator	generator
subtype-monad	monad
Undefined	null
DEFUN'd Function	internal_lambda
Array Indexing	internal_arrindexing_lazy
Assignment	internal_assignment_object

## Notes

- All type that is not `lit` only appear when the statement returns such values, e.g. `internal_lambda` only get returned by DEFUN statements as the statement itself returns defined function as well as assign them to given BASIC variable.
- As all variables will have `troType` of `lit` when they are not resolved, the property must not be used to determine the type of the variable; you must `resolve` it first.
- The type string `function` should not appear outside of TRO and `astType`; if you do see them in the wild, please check your JS code because you probably meant `usrdefun`.

## 9.3 Lambda Variables

Lambda expressions have bound variables:  $\lambda x.E$  has bound variable of  $x$  and this expression can be written in Terran BASIC as `[X] ~> E`.

However, in the creation and execution of a syntax tree, the bound variables must be able to be properly substituted, but at the same time Closure on Terran BASIC can have multiple parameters. How do we solve the substitution and name collision problem?

In 1972, dutch mathematician Nicolaas de Bruijn invented *de Bruijn Indexing*, a system we use to solve the aforementioned problems. In de Bruijn Indexing, the innermost bound variable has an index of zero\* and outer variables have greater indices. For example,  $\lambda x.\lambda y.\lambda z.x\ z(y\ z)$  is represented as  $\lambda\ \lambda\ \lambda\ 2\ 0\ (1\ 0)$

---

\*Smallest number in the set of natural numbers, which can be zero or one depending on your definition of natural numbers; we obviously chose zero.

Since closures in Terran BASIC can have multiple bound variables instead of just one as in lambda calculus, we deploy a modified version of the indexing:

*recIndex* := index of recursion depth  
*ordIndex* := index of a bound variable within a level  
*index* := array of [*recIndex*,*ordIndex*]

And consider following example code:

```
[X, Y] ~> [C] ~> ZIP(C, FILTER([M] ~> C, ZIP(X, Y)))
```

In this code, variables will have following indices:

Variable	Index	Variable	Index	Variable	Index
C (in filter of M)	[1, 0]	X	[1, 0]	M	[0, 0]
C (in outer ZIP)	[0, 0]	Y	[1, 1]		

and the program tree would look like this:




Figure 9.1: Variable Names Tree






Figure 9.2: Variable Indices Tree

As you can clearly observe, there are two s, both must refer to the same bound variable but are in different depth. In order to satisfy the constraint, two s get different

indices. If you were to change `[M]` into `[C]`, however, the inner `C` will get index of `[0, 0]`, which points to its immediate `C` and it's different `C` than one referred in the outer `ZIP`, even if the `ZIP`'s `C` has identical index of `[0, 0]`.

In `basic.js`, a variable `lambdaBoundVars` will contain such variables, and gets reused in different contexts.

# Chapter 10

## Virtual Machine

This chapter explains implementation details of Terran BASIC running on TSVM.

### 10.1 Keycodes

This is a table of keycodes recognised by the LibGDX, a framework that TSVM runs on.

Key	Code	Key	Code	Key	Code	Key	Code
1	8	+	70	v	50	F2	245
2	9	A	29	w	51	F3	246
3	10	B	30	x	52	F4	247
4	11	C	31	y	53	F5	248
5	12	D	32	z	54	F6	249
6	13	E	33	LCtrl	57	F7	250
7	14	F	34	RCtrl	58	F8	251
8	15	G	35	LShift	59	F9	252
9	16	H	36	RShift	60	F10	253
0	17	I	37	LAlt	129	F11	254
←	66	J	38	RAlt	130	Num 0	144
BkSp	67	K	39	↑	19	Num 1	145
Tab	61	L	40	↓	20	Num 2	146
`	68	M	41	←	21	Num 3	147
'	75	N	42	→	22	Num 4	148
;	43	O	43	Ins	133	Num 5	149
,	55	P	44	Del	112	Num 6	150
.	56	Q	45	PgUp	92	Num 7	151
/	76	R	46	PgDn	93	Num 8	152
[	71	S	47	Home	3	Num 9	153
]	72	T	48	End	132	NumLk	78
-	69	U	49	F1	244	*	17

Keys not listed on the table may not be available depending on the system, for example, F12 may not be recognised.

10.2 Code Page

By default TSVM uses slightly modified version of CP-437, this is a character map of it:

	00	01	02	03	04	05	06	07	08	09	0A	0B	0C	0D	0E	0F
00		☐	☐	♥	♦	♠	♣	•	■	◊	⊕	♂	♀	ℙ	♂	⊗
10	▶	◀	↑	!!	¶	§	—	‡	↑	↓	→	←	⌞	⌞	▲	▼
20		!	"	#	\$	%	&	'	(	)	*	+	,	-	.	/
30	0	1	2	3	4	5	6	7	8	9	:	;	<	=	>	?
40	Ⓟ	Ⓐ	Ⓑ	Ⓒ	Ⓓ	Ⓔ	Ⓕ	Ⓖ	Ⓗ	Ⓘ	Ⓝ	Ⓚ	Ⓛ	Ⓜ	Ⓝ	Ⓞ
50	Ⓟ	Ⓠ	Ⓡ	Ⓢ	Ⓣ	Ⓤ	Ⓥ	Ⓦ	Ⓧ	Ⓨ	Ⓩ	[	\	]	^	_
60	`	a	b	c	d	e	f	g	h	i	j	k	l	m	n	o
70	p	q	r	s	t	u	v	w	x	y	z	{		}	~	△
80	Ç	ü	é	â	ä	à	å	ç	ê	ë	è	ï	î	ì	ñ	Å
90	É	æ	œ	ô	ö	ò	û	ù	ÿ	Ö	Ü	¢	£	¥	□	⊗
A0	á	í	ó	ú	ñ	ñ	ª	º	¿	¡	¡	½	¼	¿	«	»
B0	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠
C0	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠
D0	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠	⌠
E0	α	β	Γ	π	Σ	σ	μ	γ	φ	θ	Ω	δ	ω	φ	€	Ⓜ
F0	≡	±	≥	≤	ℓ	ℓ	÷	≈	°	.	.	√	∞	∞	.	■

Figure 10.1: TSVM Character Map

10.3 Colour Palette

By default the reference graphics adapter of the TSVM uses following colour palette:




Figure 10.2: TSVM Colour Palette

0	#000F	43	#30BF	86	#624F	129	#92FF	172	#C48F	215	#F60F
1	#004F	44	#30FF	87	#628F	130	#940F	173	#C4BF	216	#F64F
2	#008F	45	#320F	88	#62BF	131	#944F	174	#C4FF	217	#F68F
3	#00BF	46	#324F	89	#62FF	132	#948F	175	#C60F	218	#F6BF
4	#00FF	47	#328F	90	#640F	133	#94BF	176	#C64F	219	#F6FF
5	#020F	48	#32BF	91	#644F	134	#94FF	177	#C68F	220	#F90F
6	#024F	49	#32FF	92	#648F	135	#960F	178	#C6BF	221	#F94F
7	#028F	50	#340F	93	#64BF	136	#964F	179	#C6FF	222	#F98F
8	#02BF	51	#344F	94	#64FF	137	#968F	180	#C90F	223	#F9BF
9	#02FF	52	#348F	95	#660F	138	#96BF	181	#C94F	224	#F9FF
10	#040F	53	#34BF	96	#664F	139	#96FF	182	#C98F	225	#FB0F
11	#044F	54	#34FF	97	#668F	140	#990F	183	#C9BF	226	#FB4F
12	#048F	55	#360F	98	#66BF	141	#994F	184	#C9FF	227	#FB8F
13	#04BF	56	#364F	99	#66FF	142	#998F	185	#CB0F	228	#FBBF
14	#04FF	57	#368F	100	#690F	143	#99BF	186	#CB4F	229	#FBBF
15	#060F	58	#36BF	101	#694F	144	#99FF	187	#CB8F	230	#FD0F
16	#064F	59	#36FF	102	#698F	145	#9B0F	188	#CBBF	231	#FD4F
17	#068F	60	#390F	103	#69BF	146	#9B4F	189	#CBFF	232	#FD8F
18	#06BF	61	#394F	104	#69FF	147	#9B8F	190	#CD0F	233	#FDBF
19	#06FF	62	#398F	105	#6B0F	148	#9BBF	191	#CD4F	234	#FDF
20	#090F	63	#39BF	106	#6B4F	149	#9BFF	192	#CD8F	235	#FF0F
21	#094F	64	#39FF	107	#6B8F	150	#9D0F	193	#CDBF	236	#FF4F
22	#098F	65	#3B0F	108	#6BBF	151	#9D4F	194	#CDFF	237	#FF8F
23	#09BF	66	#3B4F	109	#6BFF	152	#9D8F	195	#CFOF	238	#FFBF
24	#09FF	67	#3B8F	110	#6D0F	153	#9DBF	196	#CF4F	239	#FFFF
25	#0B0F	68	#3BBF	111	#6D4F	154	#9DFF	197	#CF8F	240	#000F
26	#0B4F	69	#3BFF	112	#6D8F	155	#9F0F	198	#CFBF	241	#111F
27	#0B8F	70	#3D0F	113	#6DBF	156	#9F4F	199	#CFFF	242	#222F
28	#0BBF	71	#3D4F	114	#6DFF	157	#9F8F	200	#F00F	243	#333F
29	#0BFF	72	#3D8F	115	#6F0F	158	#9FBF	201	#F04F	244	#444F
30	#0D0F	73	#3DBF	116	#6F4F	159	#9FFF	202	#F08F	245	#555F
31	#0D4F	74	#3DFF	117	#6F8F	160	#C00F	203	#F0BF	246	#666F
32	#0D8F	75	#3F0F	118	#6FBF	161	#C04F	204	#F0FF	247	#777F
33	#0DBF	76	#3F4F	119	#6FFF	162	#C08F	205	#F20F	248	#888F
34	#0DFF	77	#3F8F	120	#900F	163	#C0BF	206	#F24F	249	#999F
35	#0F0F	78	#3FBF	121	#904F	164	#C0FF	207	#F28F	250	#AAAF
36	#0F4F	79	#3FFF	122	#908F	165	#C20F	208	#F2BF	251	#BBBF
37	#0F8F	80	#600F	123	#90BF	166	#C24F	209	#F2FF	252	#CCCF
38	#0FBF	81	#604F	124	#90FF	167	#C28F	210	#F40F	253	#DDDF
39	#0FFF	82	#608F	125	#920F	168	#C2BF	211	#F44F	254	#EEEF
40	#300F	83	#60BF	126	#924F	169	#C2FF	212	#F48F	255	#0000
41	#304F	84	#60FF	127	#928F	170	#C40F	213	#F4BF		
42	#308F	85	#620F	128	#92BF	171	#C44F	214	#F4FF		

Table 10.3: Index-RGBA Table of the Colour Palette



## 10.4 MMIO

Some parts of the memory address are mapped to some devices. This section will describe useful memory addresses that can be `POKE`d.

For more detailed documentation, refer to the manuals for TSVM.

Address	Description
-1048577..-1299456	Screen Buffer
-1299457	Screen Background RED
-1299458	Screen Background GREEN
-1299459	Screen Background BLUE
-1302527..-1302528	Text Cursor Position in $row \times 80 + col$ , stored in Little Endian
-1302529..-1305088	Text Foreground Colours
-1305089..-1307648	Text Background Colours
-1307649..-1310208	Text Buffer
-1310209..-1310720	Palettes in This Pattern: 0b RRRR GGGG; 0b BBBB AAAA

## **Part III**

# **More Goodies**

## Chapter 11

# 99 Bottles of Beer

This is a sample program that prints out the infamous *99 Bottles of Beer*.

---

```
10 FOR I = 99 TO 1 STEP -1
20 MODE = 1
30 GOSUB 120
40 PRINT I;" bottle";BOTTLES;" of beer on the wall, ";i;"
   bottle";BOTTLES;" of beer."
50 MODE = 2
60 GOSUB 120
70 PRINT "Take one down and pass it around, ";(I-1);"
   bottle";BOTTLES;" of beer on the wall."
80 NEXT
90 PRINT "No more bottles of beer on the wall, no more bottles of
   beer."
100 PRINT "Go to the store and buy some more. 99 bottles of beer on
   the wall."
110 END
120 IF I == MODE THEN BOTTLES = "" ELSE BOTTLES = "s"
130 RETURN
```

---

## Chapter 12

# Amazing

This is a sample program that draw a randomised maze. The original program was on *BASIC Computer Games: Microcomputer Edition* and was translated into Terran BASIC.

---

```
1 OPTIONBASE 1
10 PRINT SPC(28);"AMAZING PROGRAM"
20 PRINT SPC(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
30 PRINT:PRINT:PRINT
100 PRINT "WHAT ARE YOUR WIDTH";:INPUT H
102 PRINT "WHAT ARE YOUR LENGTH";:INPUT V
105 IF H<>1 AND V<>1 THEN GOTO 110
106 PRINT "MEANINGLESS DIMENSIONS. TRY AGAIN.":GOTO 100
110 WS=DIM(H,V):VS=DIM(H,V)
120 PRINT:PRINT:PRINT:PRINT
160 Q=0:Z=0:X=INT(RND(1)*H+1)
165 FOR I=1 TO H
170 IF I==X THEN GOTO 173
171 PRINT ".--";
172 GOTO 175
173 PRINT ". ";
180 NEXT
190 PRINT "."
195 C=1:WS(X,1)=C:C=C+1
200 R=X:S=1:GOTO 260
210 IF R<>H THEN GOTO 240
215 IF S<>V THEN GOTO 230
220 R=1:S=1
222 GOTO 250
230 R=1:S=S+1:GOTO 250
240 R=R+1
250 IF WS(R,S)==0 THEN GOTO 210
260 IF R-1==0 THEN GOTO 530
265 IF WS(R-1,S)<>0 THEN GOTO 530
270 IF S-1==0 THEN GOTO 390
280 IF WS(R,S-1)<>0 THEN GOTO 390
290 IF R==H THEN GOTO 330
300 IF WS(R+1,S)<>0 THEN GOTO 330
310 X=INT(RND(1)*3+1)
320 ON X GOTO 790,820,860
```

```
334 IF Z==1 THEN GOTO 370
338 Q=1:GOTO 350
340 IF WS(R,S+1)<>0 THEN GOTO 370
350 X=INT(RND(1)*3+1)
360 ON X GOTO 790,820,910
370 X=INT(RND(1)*2+1)
380 ON X GOTO 790,820
390 IF R==H THEN GOTO 470
400 IF WS(R+1,S)<>0 THEN GOTO 470
405 IF S<>V THEN GOTO 420
410 IF Z==1 THEN GOTO 450
415 Q=1:GOTO 430
420 IF WS(R,S+1)<>0 THEN GOTO 450
430 X=INT(RND(1)*3+1)
440 ON X GOTO 790,860,910
450 X=INT(RND(1)*2+1)
460 ON X GOTO 790,860
470 IF S<>V THEN GOTO 490
480 IF Z==1 THEN GOTO 520
485 Q=1:GOTO 500
490 IF WS(R,S+1)<>0 THEN GOTO 520
500 X=INT(RND(1)*2+1)
510 ON X GOTO 790,910
520 GOTO 790
530 IF S-1==0 THEN GOTO 670
540 IF WS(R,S-1)<>0 THEN GOTO 670
545 IF R==H THEN GOTO 610
547 IF WS(R+1,S)<>0 THEN GOTO 610
550 IF S<>V THEN GOTO 560
552 IF Z==1 THEN GOTO 590
554 Q=1:GOTO 570
560 IF WS(R,S+1)<>0 THEN GOTO 590
570 X=INT(RND(1)*3+1)
580 ON X GOTO 820,860,910
590 X=INT(RND(1)*2+1)
600 ON X GOTO 820,860
610 IF S<>V THEN GOTO 630
620 IF Z==1 THEN GOTO 660
625 Q=1:GOTO 640
630 IF WS(R,S+1)<>0 THEN GOTO 660
640 X=INT(RND(1)*2+1)
650 ON X GOTO 820,910
660 GOTO 820
670 IF R==H THEN GOTO 740
680 IF WS(R+1,S)<>0 THEN GOTO 740
```

```
685 IF S<>V THEN GOTO 700
690 IF Z==1 THEN GOTO 730
695 Q=1:GOTO 830
700 IF WS(R,S+1)<>0 THEN GOTO 730
710 X=INT(RND(1)*2+1)
720 ON X GOTO 860,910
730 GOTO 860
740 IF S<>V THEN GOTO 760
750 IF Z==1 THEN GOTO 780
755 Q=1:GOTO 770
760 IF WS(R,S+1)<>0 THEN GOTO 780
770 GOTO 910
780 GOTO 1000
790 WS(R-1,S)=C
800 C=C+1:VS(R-1,S)=2:R=R-1
810 IF C==H*V+1 THEN GOTO 1010
815 Q=0:GOTO 260
820 WS(R,S-1)=C
830 C=C+1
840 VS(R,S-1)=1:S=S-1:IF C==H*V+1 THEN GOTO 1010
850 Q=0:GOTO 260
860 WS(R+1,S)=C
870 C=C+1:IF VS(R,S)==0 THEN GOTO 880
875 VS(R,S)=3:GOTO 890
880 VS(R,S)=2
890 R=R+1
900 IF C==H*V+1 THEN GOTO 1010
905 GOTO 530
910 IF Q==1 THEN GOTO 960
920 WS(R,S+1)=C:C=C+1:IF VS(R,S)==0 THEN GOTO 940
930 VS(R,S)=3:GOTO 950
940 VS(R,S)=1
950 S=S+1:IF C==H*V+1 THEN GOTO 1010
955 GOTO 260
960 Z=1
970 IF VS(R,S)==0 THEN GOTO 980
975 VS(R,S)=3:Q=0:GOTO 1000
980 VS(R,S)=1:Q=0:R=1:S=1:GOTO 250
1000 GOTO 210
1010 FOR J=1 TO V
1011 PRINT "|";
1012 FOR I=1 TO H
1013 IF VS(I,J)<2 THEN GOTO 1030
1020 PRINT " ";
1021 GOTO 1040
```

```
1030   PRINT " |";
1040 NEXT
1041 PRINT
1043 FOR I=1 TO H
1045   IF VS(I,J)==0 THEN GOTO 1060
1050   IF VS(I,J)==2 THEN GOTO 1060
1051   PRINT ": ";
1052   GOTO 1070
1060   PRINT "--";
1070 NEXT
1071 PRINT ". "
1072 NEXT
1073 END
```

---

## Chapter 13

# Hamurabi

This is a sample program that is the *Hamurabi* game. The original program was on *BASIC Computer Games: Microcomputer Edition* and was translated into Terran BASIC.

This game is considered as the grand ancestor of the strategy, simulation and city-building games; in fact it's so great it has got its own Wikipedia article.

---

```
10 PRINT SPC(32);"HAMURABI"
20 PRINT SPC(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
30 PRINT:PRINT:PRINT
80 PRINT "TRY YOUR HAND AT GOVERNING ANCIENT SUMERIA"
90 PRINT "FOR A TEN-YEAR TERM OF OFFICE.":PRINT
95 D1=0:P1=0
100 Z=0:P=95:S=2800:H=3000:E=H-S
110 Y=3:A=H/Y:I=5:Q=1
210 D=0
215 PRINT:PRINT:PRINT "HAMURABI: I BEG TO REPORT TO YOU,";Z=Z+1
217 PRINT "IN YEAR ";Z;",";D;" PEOPLE STARVED, ";I;" CAME TO THE
    CITY,"
220 P=P+I
227 IF Q>0 THEN GOTO 230
228 P=INT(P/2)
229 PRINT "A HORRIBLE PLAGUE STRUCK! HALF THE PEOPLE DIED."
230 PRINT "POPULATION IS NOW ";P
232 PRINT "THE CITY NOW OWNS ";A;" ACRES."
235 PRINT "YOU HARVESTED ";Y;" BUSHELS PER ACRE."
250 PRINT "THE RATS ATE ";E;" BUSHELS."
260 PRINT "YOU NOW HAVE ";S;" BUSHELS IN STORE."
261 PRINT
270 IF Z==11 THEN GOTO 860
310 C=INT(10*RND(1))
311 Y=C+17
312 PRINT "LAND IS TRADING AT ";Y;" BUSHELS PER ACRE."
320 PRINT "HOW MANY ACRES DO YOU WISH TO BUY";
321 INPUT Q
322 IF Q<0 THEN GOTO 850
323 IF Y*Q<=S THEN GOTO 330
324 GOSUB 710
325 GOTO 320
```



```

330 IF Q==0 THEN GOTO 340
331 A=A+Q:S=S-Y*Q:C=0
334 GOTO 400
340 PRINT "HOW MANY ACRES DO YOU WISH TO SELL";
341 INPUT Q
342 IF Q<0 THEN GOTO 850
343 IF Q<A THEN GOTO 350
344 GOSUB 720
345 GOTO 340
350 A=A-Q:S=S+Y*Q:C=0
400 PRINT
410 PRINT "HOW MANY BUSHEL DO YOU WISH TO FEED YOUR PEOPLE";
411 INPUT Q
412 IF Q<0 THEN GOTO 850
418 REM *** TRYING TO USE MORE GRAIN THAN IS IN SILOS?
420 IF Q<=S THEN GOTO 430
421 GOSUB 710
422 GOTO 410
430 S=S-Q:C=1:PRINT
440 PRINT "HOW MANY ACRES DO YOU WISH TO PLANT WITH SEED";
441 INPUT D
442 IF D==0 THEN GOTO 511
443 IF D<0 THEN GOTO 850
444 REM *** TRYING TO PLANT MORE ACRES THAN YOU OWN?
445 IF D<=A THEN GOTO 450
446 GOSUB 720
447 GOTO 440
449 REM *** ENOUGH GRAIN FOR SEED?
450 IF INT(D/2)<=S THEN GOTO 455
452 GOSUB 710
453 GOTO 440
454 REM *** ENOUGH PEOPLE TO TEND THE CROPS?
455 IF D<10*P THEN GOTO 510
460 PRINT "BUT YOU HAVE ONLY ";P;" PEOPLE TO TEND THE FIELDS! NOW
    THEN,"
470 GOTO 440
510 S=S-INT(D/2)
511 GOSUB 800
512 REM *** A BOUNTIFUL HARVEST!
515 Y=C:H=D*Y:E=0
521 GOSUB 800
522 IF INT(C/2)<>C/2 THEN GOTO 530
523 REM *** RATS ARE RUNNING WILD!!
525 E=INT(S/C)
530 S=S-E+H

```

```
531 GOSUB 800
532 REM *** LET'S HAVE SOME BABIES
533 I=INT(C*(20*A+S)/P/100+1)
539 REM *** HOW MANY PEOPLE HAD FULL TUMMIES?
540 C=INT(Q/20)
541 REM *** HORROS, A 15% CHANCE OF PLAGUE
542 Q=INT(10*(2*RND(1)-0.3))
550 IF P<C THEN GOTO 210
551 REM *** STARVE ENOUGH FOR IMPEACHMENT?
552 D=P-C
553 IF D>0.45*P THEN GOTO 560
554 P1=((Z-1)*P1+D*100/P)/Z
555 P=C
556 D1=D1+D
557 GOTO 215
560 PRINT
561 PRINT "YOU STARVED ";D;" PEOPLE IN ONE YEAR!!!"
565 PRINT "DUE TO THIS EXTREME MISMANAGEMENT YOU HAVE NOT ONLY"
566 PRINT "BEEN IMPEACHED AND THROWN OUT OF OFFICE BUT YOU HAVE"
567 PRINT "ALSO BEEN DECLARED NATIONAL FINK!!!"
568 GOTO 990
710 PRINT "HAMURABI: THINK AGAIN. YOU HAVE ONLY"
711 PRINT S;" BUSHELS OF GRAIN. NOW THEN,"
712 RETURN
720 PRINT "HAMURABI: THINK AGAIN. YOU OWN ONLY ";A;" ACRES. NOW THEN,"
730 RETURN
800 C=INT(RND(1)*5)+1
801 RETURN
850 PRINT
851 PRINT "HAMURABI: I CANNOT DO WHAT YOU WISH."
855 PRINT "GET YOURSELF ANOTHER STEWARD!!!!!"
857 GOTO 990
860 PRINT "IN YOUR 10-YEAR TERM OF OFFICE, ";P1;" PERCENT OF THE"
862 PRINT "POPULATION STARVED PER YEAR ON THE AVERAGE, I.E. A TOTAL
    OF"
865 PRINT D1;" PEOPLE DIED!!"
866 L=A/P
870 PRINT "YOU STARTED WITH 10 ACRES PER PERSON AND ENDED WITH"
875 PRINT L;" ACRES PER PERSON."
876 PRINT
880 IF P1>33 THEN GOTO 565
885 IF L<7 THEN GOTO 565
890 IF P1>10 THEN GOTO 940
892 IF L<9 THEN GOTO 940
895 IF P1>3 THEN GOTO 960
```

```
896 IF L<10 THEN GOTO 960
900 PRINT "A FANTASTIC PERFORMANCE!!! CHARLEMANGE, DISRAELI, AND"
905 PRINT "JEFFERSON COMBINED COULD NOT HAVE DONE BETTER!"
906 GOTO 990
940 PRINT "YOUR HEAVY-HANDED PERFORMANCE SMACKS OF NERO AND IVAN IV."
945 PRINT "THE PEOPLE (REMAINING) FIND YOU AN UNPLEASANT RULER, AND,"
950 PRINT "FRANKLY, HATE YOUR GUTS!!"
951 GOTO 990
960 PRINT "YOUR PERFORMANCE COULD HAVE BEEN SOMEWHAT BETTER, BUT"
965 PRINT "REALLY WASN'T TOO BAD AT ALL. ";INT(P*0.8*RND(1));" PEOPLE"
970 PRINT "WOULD DEARLY LIKE TO SEE YOU ASSASSINATED BUT WE ALL HAVE
    OUR"
975 PRINT "TRIVIAL PROBLEMS."
990 PRINT
991 FOR N=1 TO 10
992   EMIT 7;
993 NEXT
995 PRINT "SO LONG FOR NOW."
996 PRINT
999 END
```

---

## Chapter 14

# Hangman

This is a sample program that is the *Hangman* game. The original program was on *BASIC Computer Games: Microcomputer Edition* and was translated into Terran BASIC.

---

```
1 OPTIONBASE 1
10 PRINT SPC(32);"HANGMAN"
20 PRINT SPC(15);"CREATIVE COMPUTING MORRISTOWN, NEW JERSEY"
21 PRINT:PRINT SPC(14);"EDITOR'S NOTE: ALWAYS TYPE IN CAPITAL
    LETTERS!"
25 PRINT:PRINT
30 PSTR=DIM(12,12):LSTR=DIM(20):DSTR=DIM(20):NSTR=DIM(26):U=DIM(50)
40 C=1: N=50
50 FOR I=1 TO 20: DSTR(I)="-": NEXT: M=0
60 FOR I=1 TO 26: NSTR(I)="": NEXT
70 FOR I=1 TO 12: FOR J=1 TO 12: PSTR(I,J)=" ": NEXT: NEXT
80 FOR I=1 TO 12: PSTR(I,1)="X": NEXT
90 FOR I=1 TO 7: PSTR(1,I)="X": NEXT: PSTR(2,7)="X"
95 IF C<N THEN GOTO 100
97 PRINT "YOU DID ALL THE WORDS!!": END
100 Q=INT(N*RND(1))+1
110 IF U(Q)==1 THEN GOTO 100
115 U(Q)=1: C=C+1: RESTORE: T1=0
150 FOR I=1 TO Q: READ ASTR: NEXT
160 L=LEN(ASTR): FOR I=1 TO LEN(ASTR): LSTR(I)=MID(ASTR,I,1): NEXT
170 PRINT "HERE ARE THE LETTERS YOU USED:"
180 FOR I=1 TO 26: PRINT NSTR(I);: IF NSTR(I+1)==" " THEN GOTO 200
190 PRINT ",,": NEXT
200 PRINT: PRINT: FOR I=1 TO L: PRINT DSTR(I);: NEXT: PRINT: PRINT
210 PRINT "WHAT IS YOUR GUESS";:INPUT GSTR: R=0
220 FOR I=1 TO 26: IF NSTR(I)==" " THEN GOTO 250
230 IF GSTR==NSTR(I) THEN DO(PRINT "YOU GUESSED THAT LETTER BEFORE!";
    GOTO 170)
240 NEXT: PRINT "PROGRAM ERROR. RUN AGAIN.": END
250 NSTR(I)=GSTR: T1=T1+1
260 FOR I=1 TO L: IF LSTR(I)=GSTR THEN GOTO 280
270 NEXT: IF R==0 THEN GOTO 290
275 GOTO 300
280 DSTR(I)=GSTR: R=R+1: GOTO 270
290 M=M+1: GOTO 400
```

```

300 FOR I=1 TO L: IF DSTR(I)=="-" THEN GOTO 320
310 NEXT: GOTO 390
320 PRINT: FOR I=1 TO L: PRINT DSTR(I);: NEXT: PRINT: PRINT
330 PRINT "WHAT IS YOUR GUESS FOR THE WORD";:INPUT BSTR
340 IF ASTR==BSTR THEN GOTO 360
350 PRINT "WRONG. TRY ANOTHER LETTER.": PRINT: GOTO 170
360 PRINT "RIGHT!! IT TOOK YOU ";T1;" GUESSES!"
370 PRINT "WANT ANOTHER WORD";:INPUT WSTR: IF WSTR=="YES" THEN GOTO 50
380 PRINT: PRINT "IT'S BEEN FUN! BYE FOR NOW.": GOTO 999
390 PRINT "YOU FOUND THE WORD!": GOTO 370
400 PRINT: PRINT: PRINT"SORRY, THAT LETTER ISN'T IN THE WORD."
410 ON M GOTO 415,420,425,430,435,440,445,450,455,460
415 PRINT "FIRST, WE DRAW A HEAD": GOTO 470
420 PRINT "NOW WE DRAW A BODY.": GOTO 470
425 PRINT "NEXT WE DRAW AN ARM.": GOTO 470
430 PRINT "THIS TIME IT'S THE OTHER ARM.": GOTO 470
435 PRINT "NOW, LET'S DRAW THE RIGHT LEG.": GOTO 470
440 PRINT "THIS TIME WE DRAW THE LEFT LEG.": GOTO 470
445 PRINT "NOW WE PUT UP A HAND.": GOTO 470
450 PRINT "NEXT THE OTHER HAND.": GOTO 470
455 PRINT "NOW WE DRAW ONE FOOT": GOTO 470
460 PRINT "HERE'S THE OTHER FOOT -- YOU'RE HUNG!!"
470 ON M GOTO 480,490,500,510,520,530,540,550,560,570
480 PSTR(3,6)="-": PSTR(3,7)="-": PSTR(3,8)="-": PSTR(4,5)="(":
  PSTR(4,6)=". "
481
  PSTR(4,8)=". ":PSTR(4,9)=")":PSTR(5,6)="-":PSTR(5,7)="-":PSTR(5,8)="-":GOTO
  580
490 FOR I=6 TO 9: PSTR(I,7)="X": NEXT: GOTO 580
500 FOR I=4 TO 7: PSTR(I,I-1)="\\": NEXT: GOTO 580
510 PSTR(4,11)="/" : PSTR(5,10)="/" : PSTR(6,9)="/" : PSTR(7,8)="/" :
  GOTO 580
520 PSTR(10,6)="/" : PSTR(11,5)="/" : GOTO 580
530 PSTR(10,8)="" : PSTR(11,9)="" : GOTO 580
540 PSTR(3,11)="" : GOTO 580
550 PSTR(3,3)="/" : GOTO 580
560 PSTR(12,10)="" : PSTR(12,11)="-": GOTO 580
570 PSTR(12,3)="-": PSTR(12,4)="/"
580 FOR I=1 TO 12: FOR J=1 TO 12: PRINT PSTR(I,J);: NEXT
590 PRINT: NEXT: PRINT: PRINT: IF M<>10 THEN GOTO 170
600 PRINT "SORRY, YOU LOSE. THE WORD WAS ";ASTR
610 PRINT "YOU MISSED THAT ONE. DO YOU ";: GOTO 370
620 PRINT "TYPE YES OR NO";:INPUT YSTR: IF LEFT(YSTR,1)=="Y" THEN
  GOTO 50
700 DATA "GUM","SIN","FOR","CRY","LUG","BYE","FLY"

```

```
710 DATA "UGLY","EACH","FROM","WORK","TALK","WITH","SELF"
720 DATA "PIZZA","THING","FEIGN","FIEND","ELBOW","FAULT","DIRTY"
730 DATA "BUDGET","SPIRIT","QUAINT","MAIDEN","ESCORT","PICKAX"
740 DATA "EXAMPLE","TENSION","QUININE","KIDNEY","REPLICA","SLEEPER"
750 DATA "TRIANGLE","KANGAROO","MAHOGANY","SERGEANT","SEQUENCE"
760 DATA "MOUSTACHE","DANGEROUS","SCIENTIST","DIFFERENT","QUIESCENT"
770 DATA "MAGISTRATE","ERRONEOUSLY","LOUDSPEAKER","PHYTOTOXIC"
780 DATA "MATRIMONIAL","PARASYMPATHOMIMETIC","THIGMOTROPISM"
990 PRINT "BYE NOW"
999 END
```

---

## Chapter 15

# Plotter

This is a plotter that draws a graph of a function.

`ZEROLINE` specifies which column of the text screen is  $y = 0$ , and `AMP` specifies the Y-zoom of the plotter, line 100 controls the plotting range of  $x$ .

This example program uses sinc function as an example, specified in line 10. You can re-define the function with whatever you want, then modify line 110 to call your function i.e. `PRINT PLOTLINE(YOUR_FUNCTION_HERE, I)`.

---

```

1 ZEROLINE=10
2 AMP=20
10 DEFUN SINC(P)=IF P==0 THEN 1.0 ELSE SIN(P)/P
20 DEFUN TOCHAR(P,X)=IF (X==ROUND(ZEROLINE+P*AMP)) THEN "@" ELSE IF
    (X==ZEROLINE) THEN "|" ELSE CHR(250)
30 DEFUN SCONCAT(ACC,S)=ACC+S
40 DEFUN PLOTLINE(F,X)=FOLD(SCONCAT,"",MAP(TOCHAR~<F(X),1 TO
    ZEROLINE+AMP))
100 FOR I=-40 TO 40
110 PRINT PLOTLINE(SINC,I)
120 NEXT

```

---

## Chapter 16

**>>=**

This is proof that value-monad of the Terran BASIC obeys monad laws.

Monad laws are three equations that make sure monads to have sensible behaviours:

$$\begin{array}{ll}
 \text{return } a \gg= f & \equiv fa \\
 m \gg= \text{return} & \equiv m \\
 m \gg= (\lambda x \rightarrow k\ x \gg= h) & \equiv (m \gg= k) \gg= h
 \end{array}$$

These are referred as *Left identity*, *Right identity* and *Associativity* respectively.

---

```

10 F=[X]~>X*2 : G=[X]~>X^3 : RETN=[X]~>MRET(X)

100 PRINT:PRINT "First law: 'return a >>= k' equals to 'k a'"
110 K=[X]~>RETN(F(X)) : REM K is monad-returning function
120 A=42
130 KM=RETN(A)>>=K
140 KO=K(A)
150 PRINT("KM is ";TYPEOF(KM);", ";MJOIN(KM))
160 PRINT("KO is ";TYPEOF(KO);", ";MJOIN(KO))

200 PRINT:PRINT "Second law: 'm >>= return' equals to 'm'"
210 M=MRET(G(42))
220 MM=M>>=RETN
230 MO=M
240 PRINT("MM is ";TYPEOF(MM);", ";MJOIN(MM))
250 PRINT("MO is ";TYPEOF(MO);", ";MJOIN(MO))

300 PRINT:PRINT "Third law: 'm >>= (\x -> k x >>= h)' equals to '(m
    >>= k) >>= h'"
310 REM see line 110 for the definition of K
320 H=[X]~>RETN(G(X)) : REM H is monad-returning function
330 M=MRET(69)
340 M1=M>>=([X]~>K(X)>>=H)
350 M2=(M>>=K)>>=H
360 PRINT("M1 is ";TYPEOF(M1);", ";MJOIN(M1))
370 PRINT("M2 is ";TYPEOF(M2);", ";MJOIN(M2))

```

---

Monad laws are also preserved when arrays are used:



---

```

10 F=[X]~>RETN(X~LAST(X)*2) : G=[X]~>RETN(X~LAST(X)^3) :
    RETN=[X]~>MRET(X)

100 PRINT:PRINT "First law: 'return a >>= k' equals to 'k a'"
110 K=[X]~>F(X) : REM K is monad-returning function
120 A=42!NIL
130 KM=RETN(A)>>=K
140 KO=K(A)
150 PRINT("KM is ";TYPEOF(KM);", ", ";MJOIN(KM))
160 PRINT("KO is ";TYPEOF(KO);", ", ";MJOIN(KO))

200 PRINT:PRINT "Second law: 'm >>= return' equals to 'm'"
210 M=G(42!NIL)
220 MM=M>>=RETN
230 MO=M
240 PRINT("MM is ";TYPEOF(MM);", ", ";MJOIN(MM))
250 PRINT("MO is ";TYPEOF(MO);", ", ";MJOIN(MO))

300 PRINT:PRINT "Third law: 'm >>= (\x -> k x >>= h)' equals to '(m
    >>= k) >>= h'"
310 REM see line 110 for the definition of K
320 H=[X]~>G(X) : REM H is monad-returning function
330 M=RETN(69!NIL)
340 M1=M>>=([X]~>K(X)>>=H)
350 M2=(M>>=K)>>=H
360 PRINT("M1 is ";TYPEOF(M1);", ", ";MJOIN(M1))
370 PRINT("M2 is ";TYPEOF(M2);", ", ";MJOIN(M2))

```

---

## Chapter 17

# Bibliography

- Hagemans, Rob. 2020. “PC-BASIC Documentation.” Updated 2020-09-26 19:20:45. <https://robhagemans.github.io/pcbasic/doc/2.0/>.
- Ahl, David H and North, Steve. 1978. *BASIC Computer Games*. Microcomputer Edition. New York: Workman Pub.
- HaskellWiki. “Monad.” Updated 2020-10-20 11:05. <https://wiki.haskell.org/Monad>.
- HaskellWiki. “Monad laws.” Updated 2019-11-09 09:42. [https://wiki.haskell.org/Monad\\_laws](https://wiki.haskell.org/Monad_laws).

## Disclaimers

**O'REALLY?** Press is entirely fictional publishing entity; **O'REALLY?** Press has no affiliation whatsoever with any of the real-world publishers.

Level of humour used in this document is *super-corny*. Do not use this atrocious humour for a purpose of real-world entertainment; we take no responsibility for the consequences—losing your friends, get shunned by people, etc.

# Chapter 18

## Copyright

The source code for Terran BASIC and this documentation are distributed under the following terms:

© 2020– Minjae Song (“CuriousTorvald”)

Permission is hereby granted, free of charge, to any person obtaining a copy of this software and associated documentation files (the “Software”), to deal in the Software without restriction, including without limitation the rights to use, copy, modify, merge, publish, distribute, sublicense, and/or sell copies of the Software, and to permit persons to whom the Software is furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED “AS IS”, WITHOUT WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.

# Index

+, 24  
 −, 24  
 =, 24  
 #, 26  
 /, 24  
 \*, 24  
 \, 24  
 <\*>, 27  
 <<, 25  
 <=, 25  
 <>, 25  
 <\$>, 27  
 <~>, 27  
 ==, 25  
 =<, 25  
 =>, 25  
 ><, 25  
 >=, 25  
 >>, 25  
 >>=, 28  
 >>~, 28  
 !, 26  
 <, 25  
 >, 25  
 \, 28  
 \$, 26  
 ~>, 27  
 ., 26  
 ~<, 26  
 &, 26  
 @, 28  
 ^, 24  
 ~, 26

ABS (function), 36  
 ACO (function), 26  
 ap (operator), 27  
 apply (operator), 26  
 array (syntax), 22  
 array (type), 10  
 array operator, 26  
 ASN (function), 36  
 ATN (function), 36

BAND, 25  
 bind (operator), 28  
 bitwise operator, 25  
 boolean (type), 10  
 boolean operator, 25  
 BOR, 25  
 BXOR, 25

CBR (function), 36  
 CEIL (function), 37  
 CHR (function), 43  
 CIN (function), 39  
 CLEAR (function), 41  
 closure, 27  
 closure (tutorial), 18  
 CLPX (function), 44  
 CLS (function), 44  
 code page, 54  
 colour palette, 55  
 comparison operator, 24  
 compo (operator), 26  
 concat (operator), 26  
 cons (operator), 26  
 constants, 28  
 COS (function), 37  
 COSH (function), 37  
 curry (operator), 26  
 curry (tutorial), 18  
 curry map (operator), 27

DATA (function), 39  
 DEFUN (statement), 35  
 DGET (function), 39  
 DIM (function), 39  
 DO (function), 46

EMIT (function), 40  
 END (function), 41  
 EULER (constant), 28  
 EXP (function), 37  
 expression, 21

FILTER (function), 47

- FIX (function), 37
- FLOOR (function), 37
- FOLD (function), 47
- FOR (function), 41
- FOR-NEXT (tutorial), 13
- FOREACH (function), 41
- function (tutorial), 15
- function (type), 10
- function operators, 26
- functor (supertype), 11
- functor operators, 27
  
- generator, 26
- generator (type), 10
- GETKEYSDOWN (function), 40
- GOSUB (function), 42
- GOSUB (tutorial), 13
- GOTO (function), 42
- GOTO (tutorial), 12
- GOTOYX (function), 45
  
- HEAD (function), 43
- higher-order function (tutorial), 17
  
- ID (constant), 28
- IF (statement), 34
- INIT (function), 44
- INPUT (function), 40
- INT (function), 37
  
- join (operator), 28
  
- keycodes, 53
  
- LABEL (function), 42
- LAST (function), 44
- LEFT (function), 43
- LEN (function), 37
- line number, 21
- LOG (function), 38
  
- MAP (function), 47
- map (operator), 27
- MAP (tutorial), 17
- mathematical operator, 24
- MAX, 24
  
- MID (function), 43
- MIN, 24
- MJOIN (function), 44
- mmio, 57
- MOD, 24
- monad (tutorial), 20
- monad (type), 11
- monad operators, 27
- MRET (function), 44
  
- NEXT (function), 42
- NIL (constant), 28
- number (type), 10
- numeric literal, 22
  
- ON (statement), 34
- operator, 23
- OPTIONBASE (function), 45
- OPTIONDEBUG (function), 45
- OPTIONTRACE (function), 45
- order of precedence, 23
  
- PEEK (function), 46
- PI (constant), 28
- pipe (operator), 26
- PLOT (function), 45
- POKE (function), 46
- PRINT (function), 41
- program line, 21
- push (operator), 26
  
- READ (function), 40
- recursion (tutorial), 15
- RESTORE (function), 42
- RETURN (function), 42
- return (operator), 28
- RIGHT (function), 43
- RND (function), 38
- ROUND (function), 38
  
- sequece (operator), 28
- SGN (function), 38
- SIN (function), 38
- SINH (function), 38
- SPC (function), 43
- SQR (function), 38

- statement, 21
- STEP, 26
- string (type), 10
- string literal, 22
  
- TAIL (function), 44
- TAN (function), 39
- TANH (function), 39
- TAU (constant), 28
- TEST (function), 39
- TO, 26
- TYPEOF (function), 46
- types, 23
  
- UNDEFINED (constant), 28
- undefined (type), 11
  
- variable naming, 22

